آيين نامهء سازه های فضاکار

پیوست 2
برنامه های فرميينی برای سازه های پیوست 1
برنامهء فرميينی برای تاشه های شبيه به شکل 1

(*) Formian scheme for configurations

   similar to Fig 1:  Single layer grid with

   binate pattern. See Reference 1. (*)

L1=25;    (*) Length in U1 direction (*)

L2=25;    (*) Length in U2 direction (*)

m=10;     (*) Frequency in U1 direction (*)

n=10;      (*) Frequency in U2 direction (*)

E=rinid(m,n+1,1,1)|[0,0,0; 1,0,0]#

   rinid(m+1,n,1,1)|[0,0,0; 0,1,0];

Grid=bb(L1/m,L2/n)|E;

use &,vm(2),vt(2),vh(L1/2,-L2,3*L2,

   L1/2,L2/2,0, L1/2,L2/2,1);

clear;   draw Grid;

برنامهء فرميينی برای تاشه های شبيه به شکل 2

(*) Formian scheme for configurations

   similar to Fig 2:  Single layer grid with

   obnate pattern. See Reference 1. (*)

L1=25;  (*) Length in U1 direction (*)

L2=25;  (*) Length in U2 direction (*)

m=6;     (*) Frequency of crosses

                                     in U1 direction (*)

n=6;      (*) Frequency of crosses

                                     in U2 direction (*)

E=rinid(m,n,2,2)|lamid(1,1)|[0,0,0; 1,1,0]#

   rinid(m,2,2,2*n)|[0,0,0; 2,0,0]#

   rinid(2,n,2*m,2)|[0,0,0; 0,2,0];

Grid=bb(L1/(2*m),L2/(2*n))|E;

use &,vm(2),vt(2),vh(L1/2,-L2,3*L2,

   L1/2,L2/2,0, L1/2,L2/2,1);

clear;   draw Grid;

برنامهء فرميينی برای تاشه های شبيه به شکل 3
(*) Formian scheme for configurations

   similar to Fig 3:  Single layer grid with

   binate pattern and additional diagonal

   elements. See Reference 1. (*)

L1=25;    (*) Length in U1 direction (*)

L2=25;    (*) Length in U2 direction (*)

m=10;     (*) Frequency in U1 direction,

                                    m must be even (*)

n=10;      (*) Frequency in U2 direction (*)

E=rinid(m,n+1,1,1)|[0,0,0; 1,0,0]#

   rinid(m+1,n,1,1)|[0,0,0; 0,1,0]#

   rinid(m/2,n,2,1)|lam(1,1)|[0,1,0; 1,0,0];

Grid=bb(L1/m,L2/n)|E;

use &,vm(2),vt(2),vh(L1/2,-L2,3*L2,

   L1/2,L2/2,0, L1/2,L2/2,1);

clear;   draw Grid;

برنامهء فرميينی برای تاشه های شبيه به شکل 4
(*) Formian scheme for configurations

   similar to Fig 4:  Single layer grid with

   binate pattern and additional diagonal

   elements. See Reference 1. (*)

L1=25;    (*) Length in U1 direction (*)

L2=25;    (*) Length in U2 direction (*)

m=10;     (*) Frequency in U1 direction,

                                    m must be even (*)

n=10;      (*) Frequency in U2 direction,

                                     n must be even (*)

E=rinid(m,n+1,1,1)|[0,0,0; 1,0,0]#

   rinid(m+1,n,1,1)|[0,0,0; 0,1,0]#

   lam(2,n/2)|rinid(m/2,n/2,2,1)|lam(1,1)|

   [0,1,0; 1,0,0];

Grid=bb(L1/m,L2/n)|E;

use &,vm(2),vt(2),vh(L1/2,-L2,3*L2,

   L1/2,L2/2,0, L1/2,L2/2,1);

clear;   draw Grid;
برنامهء فرميينی برای تاشه های شبيه به شکل 5
(*) Formian scheme for configurations

   similar to Fig 5:  Single layer grid with

   binate pattern and additional diagonal

   elements. See Reference 1. (*)

L1=25;    (*) Length in U1 direction (*)

L2=25;    (*) Length in U2 direction (*)

m=10;     (*) Frequency in U1 direction,

                                    m must be even (*)

n=10;      (*) Frequency in U2 direction,

                                     n must be even (*)

E=rinid(m,n+1,1,1)|[0,0,0; 1,0,0]#

   rinid(m+1,n,1,1)|[0,0,0; 0,1,0]#

   lamid(m/2,n/2)|rinid(m/2,n/2,1,1)|

   [0,1,0; 1,0,0];

Grid=bb(L1/m,L2/n)|E;

use &,vm(2),vt(2),vh(L1/2,-L2,3*L2,

   L1/2,L2/2,0, L1/2,L2/2,1);

clear;   draw Grid;
برنامهء فرميينی برای تاشه های شبيه به شکل 6
(*) Formian scheme for configurations

   similar to Fig 6:  Single layer grid with

   obnate pattern and additional diagonal

   elements. See Reference 1. (*)

L1=25;   (*) Length in U1 direction (*)

L2=25;   (*) Length in U2 direction (*)

m=6;      (*) Frequency of crosses

                                     in U1 direction (*)

n=6;       (*) Frequency of crosses

                                     in U2 direction (*)

E=rinid(m,n,2,2)|lamid(1,1)|[0,0,0; 1,1,0]#

   rinid(m,n+1,2,2)|[0,0,0; 2,0,0]#

   rinid(2,n,2*m,2)|[0,0,0; 0,2,0];

Grid=bb(L1/(2*m),L2/(2*n))|E;

use &,vm(2),vt(2),vh(L1/2,-L2,3*L2,

   L1/2,L2/2,0, L1/2,L2/2,1);

clear;   draw Grid;
برنامهء فرميينی برای تاشه های شبيه به شکل 7
(*) Formian scheme for configurations

   similar to Fig 7:  Single layer grid with

   ternate pattern. See Reference 1. (*)

L1=25;   (*) Length in U1 direction (*)

L2=25;   (*) Length in U2 direction (*)

m=6;      (*) Frequency of crosses

                                    in U1 direction (*)

n=6;       (*) Frequency of crosses

                                    in U2 direction (*)

E=rinid(m,n,2,2)|lamid(1,1)|[0,0,0; 1,1,0]#

   rinid(2*m,2,1,2*n)|[0,0,0; 1,0,0]#

   rinid(m+1,n,2,2)|[0,0,0; 0,2,0]#

   rinid(m,n-1,2,2)|[1,1,0; 1,3,0]#

   rinid(m,2,2,2*n-1)|[1,0,0; 1,1,0];

Grid=bb(L1/(2*m),L2/(2*n))|E;

use &,vm(2),vt(2),vh(L1/2,-L2,3*L2,

   L1/2,L2/2,0, L1/2,L2/2,1);

clear;   draw Grid;
برنامهء فرميينی برای تاشه های شبيه به شکل 8
(*) Formian scheme for configurations

   similar to Fig 8:  Single layer grid with

   ternate pattern. See Reference 1. (*)

L1=25;    (*) Length in U1 direction (*)

L2=25;    (*) Length in U2 direction (*)

m=8;       (*) Frequency in U1 direction (*)

n=8;        (*) Frequency in U2 direction (*)

E=rinid(m,n+1,1,1)|[0,0,0; 1,0,0]#

   rinid(m+1,n,1,1)|[0,0,0; 0,1,0]#

   rinid(m,n,1,1)|[0,1,0; 1,0,0];

Grid=bb(L1/m,L2/n)|E;

use &,vm(2),vt(2),vh(L1/2,-L2,3*L2,

   L1/2,L2/2,0, L1/2,L2/2,1);

clear;   draw Grid;
برنامهء فرميينی برای تاشه های شبيه به شکل 9
(*) Formian scheme for configurations

   similar to Fig 9:  Single layer grid with

   ternate pattern. See Reference 1. (*)

L1=24;    (*) Length in U1 direction (*)

L2=21;    (*) Length in U2 direction (*)

m=8;       (*) Frequency in U1 direction (*)

n=8;        (*) Frequency in U2 direction,

                                       n must be even (*)

E=rinid(m,n/2,2,2)|lamid(1,1)|[0,0,0; 1,1,0]#
   rinid(m,1+n/2,2,2)|[0,0,0; 2,0,0]#

   rinid(m-1,n/2,2,2)|[1,1,0; 3,1,0]#

   rinid(2,n/2,2*m,2)|[0,0,0; 0,2,0]#

   rinid(2,n,2*m,1)|[0,0,0; 0,1,0]#

   rinid(2,n/2,2*m-1,2)|[0,1,0; 1,1,0];

Grid=bb(L1/(2*m),L2/n)|E;

use &,vm(2),vt(2),vh(L1/2,-L2,3*L2,

   L1/2,L2/2,0, L1/2,L2/2,1);

clear;   draw Grid;
برنامهء فرميينی برای تاشه های شبيه به شکل 10
(*) Formian scheme for configurations

   similar to Fig 10:  Single layer grid with

   ternate pattern and hexagonal openings.

   See Reference 1. (*)

L1=36;    (*) Length in U1 direction (*)

L2=36;    (*) Length in U2 direction (*)

m=16;      (*) Frequency in U1 direction,

            m must be a multiple of 3 plus 1 (*)

n=16;       (*) Frequency in U2 direction,

                        n must be a multiple of 4 (*)

mm=ric|((m-1)/3);  nn=ric|(N/4);

E=rinid(m,n/2,2,2)|lamid(1,1)|[0,0,0; 1,1,0]#

   rinid(m,1+n/2,2,2)|[0,0,0; 2,0,0]#

   rinid(m-1,n/2,2,2)|[1,1,0; 3,1,0]#

   rinid(2,n,2*m,1)|[0,0,0; 0,1,0]#

   rinid(2,n/2,2*m-1,2)|[0,1,0; 1,1,0];

RN=rinid(mm,nn,6,4)|[4,2,0];

Grid=bb(L1/(2*m),L2/n)|lux(RN)|E;

use &,vm(2),vt(2),vh(L1/2,-L2,3*L2,

   L1/2,L2/2,0, L1/2,L2/2,1);

clear;   draw Grid;

برنامهء فرميينی برای تاشه های شبيه به شکل 11
(*) Formian scheme for configurations

   similar to Fig 11:  Single layer grid with

   ternate pattern and hexagonal openings.
   See Reference 1. (*)

L1=36;    (*) Length in U1 direction (*)

L2=36;    (*) Length in U2 direction (*)

m=16;      (*) Frequency in U1 direction,

         m must be a multiple of 3 plus 1 (*)

n=16;       (*) Frequency in U2 direction,

          n must be a multiple of 6 plus 4 (*)

mm=ric|((m-1)/3);  nn=ric|(N/4);

E=rinid(m,n/2,2,2)|lamid(1,1)|[0,0,0; 1,1,0]#

   rinid(m,1+n/2,2,2)|[0,0,0; 2,0,0]#

   rinid(m-1,n/2,2,2)|[1,1,0; 3,1,0]#

   rinid(2,n,2*m,1)|[0,0,0; 0,1,0]#

   rinid(2,n/2,2*m-1,2)|[0,1,0; 1,1,0];

RN=rinid(mm,nn,6,6)|[4,2,0]#

   rinid(mm-1,nn-1,6,6)|[7,5,0];

Grid=bb(L1/(2*m),L2/n)|lux(RN)|E;

use &,vm(2),vt(2),vh(L1/2,-L2,3*L2,

   L1/2,L2/2,0, L1/2,L2/2,1);

clear;   draw Grid;

برنامهء فرميينی برای تاشه های شبيه به شکل 12
(*) Formian scheme for configurations

   similar to Fig 12:  Single layer grid with

   ternate pattern and triangular boundary.

   See Reference 1. (*)

L=24;  (*) Length of each side of the grid (*)

m=9;    (*) Frequency (*)

Grid=bb(L/(2*m),L*(sqrt|3)/(2*m))|

   genid(m,m,2,1,1,-1)|{[0,0,0; 2,0,0],

   [1,1,0; 0,0,0], [1,1,0; 2,0,0]};

use &,vm(2),vt(2),vh(L/2,-L,3*L,

   L/2,L/2,0, L/2,L/2,1);

clear;   draw Grid;

برنامهء فرميينی برای تاشه های شبيه 

به شکلهای 13 و 14
(*) Formian scheme for configurations

   similar to Figs 13 (n=3) and 14 (n=4):

   Single layer grid with ternate pattern 

   and triangular boundary with cut 

   corners.  See Reference 1. (*)

L=30;    (*) Length of each side of 

                   the original triangular grid (*)

m=12;   (*) Frequency of the 

                   original triangular grid (*)

n=3;      (*) Size of corner cuts (*)

E=genid(m,m,2,1,1,-1)|{[0,0,0; 2,0,0],

   [1,1,0; 0,0,0], [1,1,0; 2,0,0]};

RN=lam(1,m)|genid(n,n,2,1,1,-1)|[0,0,0]#

   genid(1,n,2,-1,-1,1)|[m,m,0];

Grid=bb(L/(2*m),L*(sqrt|3)/(2*m))|

   lux(RN)|E;

use &,vm(2),vt(2),vh(L/2,-L,3*L,

   L/2,L/2,0, L/2,L/2,1);

clear;   draw Grid;

برنامهء فرميينی برای تاشه های شبيه به شکل 15
(*) Formian scheme for configurations

   similar to Fig 15:  Single layer grid with

   ternate/hexagonal pattern and triangular

   boundary.  See Reference 1. (*)

L=36;  (*) Length of each side of the grid (*)

m=17;  (*) Frequency, m must be odd (*)

p=ric|((m-1)/2);

E=genid(m,m,2,1,1,-1)|{[0,0,0; 2,0,0],

   [1,1,0; 0,0,0], [1,1,0; 2,0,0]};

RN=genid(p,p,4,2,2,-1)|[3,1,0];

Grid=bb(L/(2*m),L*(sqrt|3)/(2*m))|

   lux(RN)|E;

use &,vm(2),vt(2),vh(L/2,-L,3*L,

   L/2,L/2,0, L/2,L/2,1);

clear;   draw Grid;

برنامهء فرميينی برای تاشه های شبيه 
به شکلهای 16 و 17
(*) Formian scheme for configurations

   similar to Figs 16 (n=5) and 17 (n=7):

   Single layer grid with ternate/hexagonal

   pattern and triangular boundary with cut

   corners.  See Reference 1. (*)

L=40;   (*) Length of each side of

                   the original triangular grid (*)

m=21;  (*) Frequency of the original

             triangular grid, m must be odd (*)

n=5;     (*) Size of corner cuts,

                                n must be odd (*)

p=ric|((m-1)/2);

E=genid(m,m,2,1,1,-1)|{[0,0,0; 2,0,0],

   [1,1,0; 0,0,0], [1,1,0; 2,0,0]};

RN=genid(p,p,4,2,2,-1)|[3,1,0]#

   lam(1,m)|genid(n,n,2,1,1,-1)|[0,0,0]#

   genid(1,n,2,-1,-1,1)|[m,m,0];

Grid=bb(L/(2*m),L*(sqrt|3)/(2*m))|

   lux(RN)|E;

use &,vm(2),vt(2),vh(L/2,-L,3*L,

   L/2,L/2,0, L/2,L/2,1);

clear;   draw Grid;

برنامهء فرميينی برای تاشه های شبيه 

به شکلهای 18، 19 و 20
(*) Formian scheme for configurations

   similar to Fig 18 (m=6, n=5), Fig 19
   (m=5, n=8) and Fig 20 (m=4, n=11):

   Sectorate single layer grid with ternate

   pattern. See Reference 1. (*)

L=20;  (*) Length of each side of the grid (*)

m=6;    (*) Frequency (*)

n=5;     (*) Number of sides (*)

E=bb(L/(2*m),L/(2*m*tan|(180/n)))|

   genid(1,m,2,1,-1,1)|{[0,0,0; -1,1,0],

   [0,0,0; 1,1,0], [-1,1,0; 1,1,0]};

Grid=pex|rosad(0,0,n,360/n)|E;

use &,vm(2),vt(2),vh(0,-2*L,5*L,

   0,0,0, 0,0,1);

clear;   draw Grid;

برنامهء فرميينی برای تاشه های شبيه 

به شکلهای 21، 22 و 23
(*) Formian scheme for configurations

   similar to Fig 21 (m=13, n=5), 22
   (m=11, n=7) and Fig 23 (m=10, n=7):

   Sectorate single layer grid with

   ternate/hexagonal pattern.

   See Reference 1. (*)

L=20;  (*) Length of each side of the grid (*)

m=13;  (*) Frequency of the 

                   original ternate grid (*)

n=5;     (*) Number of sides (*)

E=genid(1,m,2,1,-1,1)|{[0,0,0; -1,1,0],

   [0,0,0; 1,1,0], [-1,1,0; 1,1,0]};

RN=genid(1,ric|((m+1)/2),4,2,-2,1)|[0,0,0];

Grid=rosad(0,0,n,360/n)|bb(L/(2*m),

   L/(2*m*tan|(180/n)))|lux(RN)|E;

use &,vm(2),vt(2),vh(0,-2*L,5*L,

   0,0,0, 0,0,1);

clear;   draw Grid;

برنامهء فرميينی برای تاشه های شبيه 

به شکلهای 24 و 25
(*) Formian scheme for configurations

   similar to Fig 24 (m=9, n=6) and

   Fig 25 (m=9, n=9):  Sectorate single

   layer grid with ternate/hexagonal pattern.

   See Reference 1. (*)

L=20;  (*) Length of each side of the grid (*)

m=9;   (*) Frequency of the original ternate

                grid, m must be a multiple of 3 (*)

n=6;     (*) Number of sides (*)

E=genid(1,m,2,1,-1,1)|{[0,0,0; -1,1,0],

   [0,0,0; 1,1,0], [-1,1,0; 1,1,0]};

RN=genid(1,((m/3)+1),6,3,-3,1)|[0,0,0];

Grid=pex|rosad(0,0,n,360/n)|bb(L/(2*m),

   L/(2*m*tan|(180/n)))|lux(RN)|E;

use &,vm(2),vt(2),vh(0,-2*L,5*L,

   0,0,0, 0,0,1);

clear;   draw Grid;

برنامهء فرميينی برای تاشه های شبيه 

به شکلهای 26 و 27
(*) Formian scheme for configurations

   similar to Fig 26 (m=11, n=5) and

   Fig 27 (m=11, n=6):  Sectorate single

   layer grid with ternate/hexagonal pattern.

   See Reference 1. (*)

L=20;  (*) Length of each side of the grid (*)

m=11;  (*) Frequency of the original ternate 

    grid, m must be a multiple of 3 minus 1(*)

n=5;     (*) Number of sides (*)

q=ric|((m+1)/3);

E=[-1,1,0; 1,1,0]#lam(1,0)|{[-1,1,0; 0,2,0],

   [-2,2,0; 0,2,0]};

F=genid(1,q,6,3,-3,1)|E#

   genid(1,q-1,6,3,-3,1)|(lamid(0,3)|lam(1,-1)|

   [-2,2,0; -1,3,0]#ref(2,3)|E);

Grid=rosad(0,0,n,360/n)|bb(L/(2*m),

   L/(2*m*tan|(180/n)))|F;

use &,vm(2),vt(2),vh(0,-2*L,5*L,

   0,0,0, 0,0,1);

clear;   draw Grid;

برنامهء فرميينی برای تاشه های شبيه 

به شکلهای 28، 29 و 30
(*) Formian scheme for configurations

   similar to Fig 28 (n=6), Fig 29 (n=8)

   and Fig 30 (n=11):  Sectorate single layer

   grid with hexagonal pattern.

   See Reference 1. (*)

L=20;  (*) Length of each side of the grid (*)

m=9;    (*) Frequency of the original ternate

                grid, m must be a multiple of 3 (*)

n=6;     (*) Number of sides (*)

E=genid(1,m,2,1,-1,1)|{[0,0,0; -1,1,0],

   [0,0,0; 1,1,0], [-1,1,0; 1,1,0]};

RN=genid(1,((m/3)+1),6,3,-3,1)|[0,0,0]#

   genid(1,m/3,6,3,-3,1)|[0,2,0]#

   genid(1,((m/3)-1),6,3,-3,1)|[0,4,0];

Grid=pex|rosad(0,0,n,360/n)|bb(L/(2*m),

   L/(2*m*tan|(180/n)))|lux(RN)|E;

use &,vm(2),vt(2),vh(0,-2*L,5*L,

   0,0,0, 0,0,1);

clear;   draw Grid;

برنامهء فرميينی برای تاشه های شبيه به شکل 31
(*) Formian scheme for configurations

   similar to Fig 31:  Single layer grid with

   hexagonal (honycomb) pattern.
   See Reference 1. (*)

L1=25;   (*) Length in U1 direction (*)

L2=24;   (*) Length in U2 direction (*)

m=4;      (*) Frequency of second row

               of hegxagons in U1 direction (*)

n=5;       (*) Frequency of hegxagons in
                U2 direction along the sides (*)

E=rinid(m,n,6,2)|lamid(2,2)|[1,1,0; 0,2,0]#
   rinid(m,n+1,6,2)|[1,1,0; 3,1,0]#
   rin(1,m-1,6)|lamid(5,n+1)|[4,0,0; 3,1,0]#

   rinid(m-1,n+2,6,2)|[4,0,0; 6,0,0];

Grid=bb(L1/(6*m-2),L2/(2*(n+1)))|E;

use &,vm(2),vt(2),vh(L1/2,-L2,3*L2,

   L1/2,L2/2,0, L1/2,L2/2,1);

clear;   draw Grid;
برنامهء فرميينی برای تاشه های شبيه به شکل 32
(*) Formian scheme for configurations

   similar to Fig 32:  Single layer grid with

   quadnate pattern. See Reference 1. (*)

L1=25;  (*) Length in U1 direction (*)

L2=25;  (*) Length in U2 direction (*)

m=6;     (*) Frequency of crosses

                                     in U1 direction (*)

n=6;      (*) Frequency of crosses

                                     in U2 direction (*)

E=rinid(m,n,2,2)|lamid(1,1)|[0,0,0; 1,1,0]#

   rinid(m,n+1,2,2)|[0,0,0; 2,0,0]#

   rinid(m+1,n,2,2)|[0,0,0; 0,2,0];

Grid=bb(L1/(2*m),L2/(2*n))|E;

use &,vm(2),vt(2),vh(L1/2,-L2,3*L2,

   L1/2,L2/2,0, L1/2,L2/2,1);

clear;   draw Grid;
برنامهء فرميينی برای تاشه های شبيه به شکل 33
(*) Formian scheme for configurations

   similar to Fig 33:  Single layer grid with

   quadnate pattern and openings.

   See Reference 1. (*)

L1=25;  (*) Length in U1 direction (*)

L2=25;  (*) Length in U2 direction (*)

m=13;     (*) Frequency of crosses in U1

                direction, m must be a multiple

                of 3 plus 1 (*)

n=13;     (*) Frequency of crosses in U2

                direction, n must be a multiple

                of 3 plus 1 (*)

E=rinid(m,n,2,2)|lamid(1,1)|[0,0,0; 1,1,0]#

   rinid(m,n+1,2,2)|[0,0,0; 2,0,0]#

   rinid(m+1,n,2,2)|[0,0,0; 0,2,0];

RN=rinid((m-1)/3,(n-1)/3,6,6)|[4,4,0];

Grid=bb(L1/(2*m),L2/(2*n))|lux(RN)|E;

use &,vm(2),vt(2),vh(L1/2,-L2,3*L2,

   L1/2,L2/2,0, L1/2,L2/2,1);

clear;   draw Grid;

برنامهء فرميينی برای تاشه های شبيه به شکل 34
(*) Formian scheme for configurations

   similar to Fig 34:  Single layer grid with

   quadnate pattern and openings.

   See Reference 1. (*)

L1=25;  (*) Length in U1 direction (*)

L2=25;  (*) Length in U2 direction (*)

m=12;    (*) Frequency of crosses in U1

     direction, m must be a multiple of 4 (*)

n=12;     (*) Frequency of crosses in U2

      direction, n must be a multiple of 4 (*)

E=rinid(m,n,2,2)|lamid(1,1)|[0,0,0; 1,1,0]#

   rinid(m,n+1,2,2)|[0,0,0; 2,0,0]#

   rinid(m+1,n,2,2)|[0,0,0; 0,2,0];

RN=rinid(m/4,n/4,8,8)|[4,4,0]#

   rinid((m/4)-1,(n/4)-1,8,8)|[8,8,0];

Grid=bb(L1/(2*m),L2/(2*n))|lux(RN)|E;

use &,vm(2),vt(2),vh(L1/2,-L2,3*L2,

   L1/2,L2/2,0, L1/2,L2/2,1);

clear;   draw Grid;

برنامهء فرميينی برای تاشه های شبيه به شکل 35
(*) Formian scheme for configurations

   similar to Fig 35:  Single layer grid with

   quadnate pattern. See Reference 1. (*)

L1=25;  (*) Length in U1 direction (*)

L2=25;  (*) Length in U2 direction (*)

m=5;     (*) Frequency of crosses

                                     in U1 direction (*)

n=5;      (*) Frequency of crosses

                                     in U2 direction (*)

E=rinid(m,n,2,2)|lamid(1,1)|[0,0,0; 1,1,0]#

   rinid(2*m,2*n+1,1,1)|[0,0,0; 1,0,0]#

   rinid(2*m+1,2*n,1,1)|[0,0,0; 0,1,0];

Grid=bb(L1/(2*m),L2/(2*n))|E;

use &,vm(2),vt(2),vh(L1/2,-L2,3*L2,

   L1/2,L2/2,0, L1/2,L2/2,1);

clear;   draw Grid;
برنامهء فرميينی برای تاشه های شبيه به شکل 36
(*) Formian scheme for configurations

   similar to Fig 36:  Single layer grid with

   quadnate pattern with openings.

   See Reference 1. (*)

L1=25;  (*) Length in U1 direction (*)

L2=25;  (*) Length in U2 direction (*)

m=7;     (*) Frequency of crosses in U1

                   direction, m must be odd (*)

n=7;      (*) Frequency of crosses in U2

                   direction, n must be odd (*)

E=rinid(m,n,2,2)|lamid(1,1)|[0,0,0; 1,1,0]#

   rinid(2*m,2*n+1,1,1)|[0,0,0; 1,0,0]#

   rinid(2*m+1,2*n,1,1)|[0,0,0; 0,1,0];

RN=rinid((m-1)/2,(n-1)/2,4,4)|[3,3,0];

Grid=bb(L1/(2*m),L2/(2*n))|

   (lux(RN)|E#rinid((m-1)/2,(n-1)/2,4,4)|

   rosad(3,3)|[3,2,0; 2,3,0]);

use &,vm(2),vt(2),vh(L1/2,-L2,3*L2,

   L1/2,L2/2,0, L1/2,L2/2,1);

clear;   draw Grid;

برنامهء فرميينی برای تاشه های شبيه به شکل 37
(*) Formian scheme for configurations

   similar to Fig 37:  Double layer grid with

   binate on binate pattern. See Section 1.7

   of Reference 1. (*)

L1=20;    (*) Length in U1 direction (*)

L2=20;    (*) Length in U2 direction (*)

D=2;       (*) Depth of grid (*)

m=6;       (*) Frequency in U1 direction (*)

n=6;        (*) Frequency in U2 direction (*)

T=rinid(m,n+1,2,2)|[0,0,1; 2,0,1]#

   rinid(m+1,n,2,2)|[0,0,1; 0,2,1];

B=rinid(m-1,n,2,2)|[1,1,0; 3,1,0]#

   rinid(m,n-1,2,2)|[1,1,0; 1,3,0];

W=rinid(m,n,2,2)|lamid(1,1)|[0,0,1; 1,1,0];

Grid=bt(L1/(2*m),L2/(2*n),D)|(T#B#W);

use &,vm(2),vt(2),vh(L1/2,-L2/3,2*L2,

   L1/2,L2/2,0, L1/2,L2/2,1);

clear;   draw Grid;
برنامهء فرميينی برای تاشه های شبيه به شکل 38
(*) Formian scheme for configurations

  similar to Fig 38:  Double layer grid with

  binate on binate pattern and some openings.

  See Section 1.7 of Reference 1. (*)

L1=20;    (*) Length in U1 direction (*)

L2=20;    (*) Length in U2 direction (*)

D=2;        (*) Depth of grid (*)

m=7;        (*) Frequency in

             U1 direction, m must be odd (*)

n=7;         (*) Frequency in

             U2 direction, n must be odd (*)

T=rinid(m,n+1,2,2)|[0,0,1; 2,0,1]#

   rinid(m+1,n,2,2)|[0,0,1; 0,2,1];

B=rinid(m-1,n,2,2)|[1,1,0; 3,1,0]#

   rinid(m,n-1,2,2)|[1,1,0; 1,3,0];

W=rinid(m,n,2,2)|lamid(1,1)|[0,0,1; 1,1,0];

RN=lamid(m,n)|[m-2,n-2,0];

G=T#lux(RN)|(B#W);

Grid=bt(L1/(2*m),L2/(2*n),D)|G;

use &,vm(2),vt(2),vh(L1/2,-L2/6,2*L2,

   L1/2,L2/2,0, L1/2,L2/2,1);

clear;   draw Grid;
برنامهء فرميينی برای تاشه های شبيه به شکل 39
(*) Formian scheme for configurations

   similar to Fig 39:  Double layer grid with

   binate on larger binate pattern.

   See Section 1.7 of Reference 1. (*)

L1=20;    (*) Length in U1 direction (*)

L2=20;    (*) Length in U2 direction (*)

D=2;        (*) Depth of grid (*)

m=7;        (*) Frequency in

             U1 direction, m must be odd (*)

n=7;         (*) Frequency in

             U2 direction, n must be odd (*)

T=rinid(m,n+1,2,2)|[0,0,1; 2,0,1]#

   rinid(m+1,n,2,2)|[0,0,1; 0,2,1];

B=rinid(m-1,n,2,2)|[1,1,0; 3,1,0]#

   rinid(m,n-1,2,2)|[1,1,0; 1,3,0];

W=rinid(m,n,2,2)|lamid(1,1)|[0,0,1; 1,1,0];

RN=rinid(tic|(m/2),tic|(n/2),4,4)|[3,3,0];

G=T#lux(RN)|(B#W);

Grid=bt(L1/(2*m),L2/(2*n),D)|G;

use &,vm(2),vt(2),vh(L1/2,-L2/6,2*L2,

   L1/2,L2/2,0, L1/2,L2/2,1);

clear;   draw Grid;
برنامهء فرميينی برای تاشه های شبيه به شکل 40
(*) Formian scheme for configurations

   similar to Fig 40:  Mansard double layer

   grid with binate on binate pattern.

   See Section 1.7 of Reference 1. (*)

L1=20;    (*) Length in U1 direction (*)

L2=20;    (*) Length in U2 direction (*)

D=2;       (*) Depth of grid (*)

m=6;       (*) Frequency in U1 direction (*)

n=6;        (*) Frequency in U2 direction (*)

T=rinid(m-1,n,2,2)|[1,1,1; 3,1,1]#

   rinid(m,n-1,2,2)|[1,1,1; 1,3,1];

B=rinid(m,n+1,2,2)|[0,0,0; 2,0,0]#

   rinid(m+1,n,2,2)|[0,0,0; 0,2,0];

W=rinid(m,n,2,2)|lamid(1,1)|[0,0,0; 1,1,1];

Grid=bt(L1/(2*m),L2/(2*n),D)|(T#B#W);

use &,vm(2),vt(2),vh(L1/2,-L2/3,2*L2,

   L1/2,L2/2,0, L1/2,L2/2,1);

clear;   draw Grid;

برنامهء فرميينی برای تاشه های شبيه به شکل 41

(*) Formian scheme for configurations

   similar to Fig 41:  Mansard double layer

   grid with binate on binate pattern and

   some openings.

   See Section 1.7 of Reference 1. (*)

L1=24;    (*) Length in U1 direction (*)

L2=24;    (*) Length in U2 direction (*)

D=2;        (*) Depth of grid (*)

m=8;        (*) Frequency in

             U1 direction, m must be even (*)

n=8;         (*) Frequency in

             U2 direction, n must be even (*)

T=rinid(m-1,n,2,2)|[1,1,1; 3,1,1]#

   rinid(m,n-1,2,2)|[1,1,1; 1,3,1];

B=rinid(m,n+1,2,2)|[0,0,0; 2,0,0]#

   rinid(m+1,n,2,2)|[0,0,0; 0,2,0];

W=rinid(m,n,2,2)|lamid(1,1)|[0,0,0; 1,1,1];

RN=lamid(m,n)|[m-4,n-4,0]#[m,n,0];
G=T#lux(RN)|(B#W);

Grid=bt(L1/(2*m),L2/(2*n),D)|G;

use &,vm(2),vt(2),vh(L1/2,-L2/6,3*L2,

   L1/2,L2/2,0, L1/2,L2/2,1);

clear;   draw Grid;
برنامهء فرميينی برای تاشه های شبيه به شکل 42
(*) Formian scheme for configurations

   similar to Fig 42:  Mansard double layer

   grid with binate on larger binate pattern.

   See Section 1.7 of Reference 1. (*)

L1=24;    (*) Length in U1 direction (*)

L2=24;    (*) Length in U2 direction (*)

D=2;        (*) Depth of grid (*)

m=8;        (*) Frequency in

             U1 direction, m must be even (*)

n=8;         (*) Frequency in

             U2 direction, n must be even (*)

T=rinid(m-1,n,2,2)|[1,1,1; 3,1,1]#

   rinid(m,n-1,2,2)|[1,1,1; 1,3,1];

B=rinid(m,n+1,2,2)|[0,0,0; 2,0,0]#

   rinid(m+1,n,2,2)|[0,0,0; 0,2,0];

W=rinid(m,n,2,2)|lamid(1,1)|[0,0,0; 1,1,1];

RN=rinid((m-2)/2,(n-2)/2,4,4)|[4,4,0];

G=T#lux(RN)|(B#W);

Grid=bt(L1/(2*m),L2/(2*n),D)|G;

use &,vm(2),vt(2),vh(L1/2,-L2/6,3*L2,

   L1/2,L2/2,0, L1/2,L2/2,1);

clear;   draw Grid;
برنامهء فرميينی برای تاشه های شبيه به شکل 43
(*) Formian scheme for configurations

   similar to Fig 43:  Double layer grid with

   obnate on obnate pattern.
   See Reference 1 (*)

L1=20;    (*) Length in U1 direction (*)

L2=20;    (*) Length in U2 direction (*)

D=2;       (*) Depth of grid (*)

m=5;       (*) Frequency in U1 direction (*)

n=5;        (*) Frequency in U2 direction (*)

T=rinid(m,n,2,2)|lamid(1,1)|[1,0,1; 0,1,1]#

   rinid(m-1,2,2,2*n)|[1,0,1; 3,0,1]#

   rinid(2,n-1,2*m,2)|[0,1,1; 0,3,1];

B=rinid(m-1,n-1,2,2)|lamid(2,2)|
   [1,1,0; 2,2,0]#rinid(m-1,2,2,2*(n-1))|

   [1,1,0; 3,1,0]#rinid(2,n-1,2*(m-1),2)|

   [1,1,0; 1,3,0];

W=rinid(m,n,2,2)|rosad(1,1)|[1,0,1; 1,1,0]#

   rinid(m-1,n-1,2,2)|rosad(2,2)|
   [2,1,1; 2,2,0];

Grid=bt(L1/(2*m),L2/(2*n),D)|(T#B#W);

use &,vm(2),vt(2),vh(L1/2,-L2/3,2*L2,

   L1/2,L2/2,0, L1/2,L2/2,1);

clear;   draw Grid;
برنامهء فرميينی برای تاشه های شبيه به شکل 44
(*) Formian scheme for configurations

   similar to Fig 44:  Double layer grid with

   obnate on larger obnate pattern.

   See Reference 1 (*)

L1=20;    (*) Length in U1 direction (*)

L2=20;    (*) Length in U2 direction (*)

D=2;       (*) Depth of grid (*)

m=6;       (*) Frequency in

            U1 direction, m must be even (*)

n=6;        (*) Frequency in

            U2 direction, n must be even (*)

T=rinid(m,n,2,2)|lamid(1,1)|[1,0,1; 0,1,1]#

   rinid(m-1,2,2,2*n)|[1,0,1; 3,0,1]#

   rinid(2,n-1,2*m,2)|[0,1,1; 0,3,1];

B=rinid(m-1,n-1,2,2)|lamid(2,2)|
   [1,1,0; 2,2,0]#rinid(m-1,2,2,2*(n-1))|

   [1,1,0; 3,1,0]#rinid(2,n-1,2*(m-1),2)|

   [1,1,0; 1,3,0];

W=rinid(m,n,2,2)|rosad(1,1)|

   [1,0,1; 1,1,0]#rinid(m-1,n-1,2,2)|

   rosad(2,2)|[2,1,1; 2,2,0];

RN=rinid(ric|((m-2)/2),ric|((n-2)/2),4,4)|

   rosad(4,4)|[4,2,0];

G=T#lux(RN)|(B#W);

Grid=bt(L1/(2*m),L2/(2*n),D)|G;

use &,vm(2),vt(2),vh(L1/2,-L2/5,2*L2,

   L1/2,L2/2,0, L1/2,L2/2,1);

clear;   draw Grid;
برنامهء فرميينی برای تاشه های شبيه به شکل 45
(*) Formian scheme for configurations

   similar to Fig 45:  Mansard double layer

   grid with obnate on obnate pattern.
   See Reference 1 (*)

L1=20;    (*) Length in U1 direction (*)

L2=20;    (*) Length in U2 direction (*)

D=2;       (*) Depth of grid (*)

m=5;       (*) Frequency in U1 direction (*)

n=5;        (*) Frequency in U2 direction (*)

T=rinid(m-1,n-1,2,2)|lamid(2,2)|

   [1,1,1; 2,2,1]#rinid(m-1,2,2,2*(n-1))|

   [1,1,1; 3,1,1]#rinid(2,n-1,2*(m-1),2)|

   [1,1,1; 1,3,1];

B=rinid(m,n,2,2)|lamid(1,1)|[1,0,0; 0,1,0]#

   rinid(m-1,2,2,2*n)|[1,0,0; 3,0,0]#

   rinid(2,n-1,2*m,2)|[0,1,0; 0,3,0];

W=rinid(m,n,2,2)|rosad(1,1)|[1,0,0; 1,1,1]#

   rinid(m-1,n-1,2,2)|

   rosad(2,2)|[2,1,0; 2,2,1];

Grid=bt(L1/(2*m),L2/(2*n),D)|(T#B#W);

use &,vm(2),vt(2),vh(L1/2,-L2/3,2*L2,

   L1/2,L2/2,0, L1/2,L2/2,1);

clear;   draw Grid;
برنامهء فرميينی برای تاشه های شبيه به شکل 46
(*) Formian scheme for configurations

   similar to Fig 46:  Mansard double layer

   grid with obnate on larger obnate pattern.

   See Reference 1 (*)

L1=20;    (*) Length in U1 direction (*)

L2=20;    (*) Length in U2 direction (*)

D=2;       (*) Depth of grid (*)

m=7;       (*) Frequency in U1 direction (*)

n=7;        (*) Frequency in U2 direction (*)

T=rinid(m-1,n-1,2,2)|lamid(2,2)|

   [1,1,1; 2,2,1]#rinid(m-1,2,2,2*(n-1))|

   [1,1,1; 3,1,1]#rinid(2,n-1,2*(m-1),2)|

   [1,1,1; 1,3,1];

B=rinid(m,n,2,2)|lamid(1,1)|[1,0,0; 0,1,0]#

   rinid(m-1,2,2,2*n)|[1,0,0; 3,0,0]#

   rinid(2,n-1,2*m,2)|[0,1,0; 0,3,0];

W=rinid(m,n,2,2)|rosad(1,1)|[1,0,0; 1,1,1]#

   rinid(m-1,n-1,2,2)|

   rosad(2,2)|[2,1,0; 2,2,1];

RN={[m,n-3,0], [m-3,n,0], [m,n+3,0],
   [m+3,n,0]};

G=T#lux(RN)|(B#W);

Grid=bt(L1/(2*m),L2/(2*n),D)|G;

use &,vm(2),vt(2),vh(L1/2,-L2/5,2*L2,

   L1/2,L2/2,0, L1/2,L2/2,1);

clear;   draw Grid;
برنامهء فرميينی برای تاشه های شبيه به شکل 47
(*) Formian scheme for configurations

   similar to Fig 47:  Double layer grid with

   binate on larger obnate pattern.

   See Reference 1 (*)

L1=20;    (*) Length in U1 direction (*)

L2=20;    (*) Length in U2 direction (*)

D=2;       (*) Depth of grid (*)

m=7;       (*) Frequency in

            U1 direction, m must be odd (*)

n=7;        (*) Frequency in

            U2 direction, n must be odd (*)

T=rinid(m,n+1,2,2)|[0,0,1; 2,0,1]#

   rinid(m+1,n,2,2)|[0,0,1; 0,2,1];

B=rinid(tic|(m/2),tic|(n/2),4,4)|lamid(3,3)|

   [1,1,0; 3,3,0]#rinid(m-1,2,2,2*(n-1))|

   [1,1,0; 3,1,0]#rinid(2,n-1,2*(m-1),2)|

   [1,1,0; 1,3,0];

RN=rinid(ric|((m-3)/2),ric|((n-1)/2),4,4)|

   [5,3,0]#rinid(ric|((m-1)/2),ric|

   ((n-3)/2),4,4)|[3,5,0];

W=lux(RN)|rinid(m,n,2,2)|

   lamid(1,1)|[0,0,1; 1,1,0];

Grid=bt(L1/(2*m),L2/(2*n),D)|(T#B#W);

use &,vm(2),vt(2),vh(L1/2,-L2/2,3*L2,

   L1/2,L2/2,0, L1/2,L2/2,1);

clear;   draw Grid;

برنامهء فرميينی برای تاشه های شبيه به شکل 48
(*) Formian scheme for configurations

   similar to Fig 48:  Double layer grid with

   obnate on larger binate pattern.

   See Reference 1 (*)

L1=20;    (*) Length in U1 direction (*)

L2=20;    (*) Length in U2 direction (*)

D=2;       (*) Depth of grid (*)

m=5;       (*) Frequency in

            U1 direction, m must be odd (*)

n=5;        (*) Frequency in

            U2 direction, n must be odd (*)

T=rinid(m,n,2,2)|lamid(1,1)|[1,0,1; 0,1,1]#

   rinid(m-1,2,2,2*n)|[1,0,1; 3,0,1]#

   rinid(2,n-1,2*m,2)|[0,1,1; 0,3,1];

B=rinid(m-1,n,2,2)|[1,1,0; 3,1,0]#

   rinid(m,n-1,2,2)|[1,1,0; 1,3,0];

W=rinid(m,n,2,2)|rosad(1,1)|[1,0,1; 1,1,0];

Grid=bt(L1/(2*m),L2/(2*n),D)|(T#B#W);

use &,vm(2),vt(2),vh(L1/2,-L2/2,3*L2,

   L1/2,L2/2,0, L1/2,L2/2,1);

clear;   draw Grid;
برنامهء فرميينی برای تاشه های شبيه به شکل 49
(*) Formian scheme for configurations

   similar to Fig 49:  Double layer grid with

   ternate on ternate pattern.

   See Reference 1 (*)

L1=24;    (*) Length in U1 direction (*)

L2=22;    (*) Length in U2 direction (*)

D=2;       (*) Depth of grid (*)

m=5;       (*) Frequency in U1 direction (*)

n=6;        (*) Frequency in

                 U2 direction, n must be even (*)
T=rinid(m,n/2,2,6)|lamid(2,3)|[1,0,1; 2,3,1]#
   rinid(m,1+n/2,2,6)|[1,0,1; 3,0,1]#

   rinid(m-1,n/2,2,6)|[2,3,1; 4,3,1];

B=rinid(m+1,n/2,2,6)|lamid(1,4)|
   [0,1,0; 1,4,0]#rinid(m+1,1+n/2,2,6)|
   [0,1,0; 2,1,0]#rinid(m,n/2,2,6)|

   [1,4,0; 3,4,0];
W=rinid(m+1,1+n/2,2,6)|lam(1,1)|

   [0,1,0; 1,0,1]#rinid(m,n/2,2,6)|

   lam(1,2)|[1,4,0; 2,3,1]#

   rinid(m+1,n/2,2,6)|[1,4,0; 1,6,1]#
   rinid(m,n/2,2,6)|[2,1,0; 2,3,1];

Grid=bt(L1/(2*m+2),L2/(3*n+1),D)|
   (T#B#W);

use &,vm(2),vt(2),vh(L1/2,-L2/2,4*L2,

   L1/2,L2/2,0, L1/2,L2/2,1);

clear;   draw Grid;
برنامهء فرميينی برای تاشه های شبيه به شکل 50
(*) Formian scheme for configurations

   similar to Fig 50:  Double layer grid with

   hexagon on hexagon pattern.

   See Reference 1 (*)
L1=22;    (*) Length in U1 direction (*)

L2=20;    (*) Length in U2 direction (*)

D=2;       (*) Depth of grid (*)

m=5;       (*) Frequency in U1 direction (*)

n=3;        (*) Frequency in U2 direction (*)

T=rinid(m+1,n,4,12)|lamid(1,6)|

   [0,3,1; 1,6,1]#rinid(m,n,4,12)|lamid(3,6)|

   [2,3,1; 3,0,1]#rinid(2*m+1,2*n,2,6)|

   [0,3,1; 2,3,1];

B=rinid(m,n,4,12)|lamid(3,6)|[1,4,0; 3,2,0]#

   rinid(m+1,n,4,12)|[1,4,0; 1,8,0]#

   rinid(m,n-1,4,12)|[3,10,0; 3,14,0];

W=rinid(m+1,n,4,12)|(lamid(1,6)|
   [0,3,1; 1,4,0]#lam(2,6)|[1,6,1; 1,4,0])#
   rinid(m,n,4,12)|(lamid(3,6)|[2,3,1; 3,2,0]#

   lam(2,6)|[3,0,1; 3,2,0]);

Grid=bt(L1/(4*m+2),L2/(12*n),D)|
   (T#B#W);

use &,vm(2),vt(2),vh(L1/2,-L2,6*L2,

   L1/2,L2/2,0, L1/2,L2/2,1);

clear;   draw Grid;
برنامهء فرميينی برای تاشه های شبيه به شکل 51
(*) Formian scheme for configurations

   similar to Fig 51:  Double layer grid with

   hexagon on triangle pattern.

   See Reference 1 (*)

L1=22;   (*) Length in U1 direction (*)

L2=20;   (*) Length in U2 direction (*)

D=2;      (*) Depth of grid (*)

m=4;      (*) Frequency in U1 direction,

                    number of hexagons in the

                    second row of top layer (*)

n=5;       (*) Frequency in U2 direction,

                    number of hexagons along

                    the sides of top layer (*)

T=rinid(m,n,6,2)|lamid(2,2)|[1,1,1; 0,2,1]#

   rin(1,m-1,6)|lamid(5,n+1)|[3,1,1; 4,0,1]#

   rinid(m,n+1,6,2)|[1,1,1; 3,1,1]#
   rinid(m-1,n+2,6,2)|[4,0,1; 6,0,1];

B=rinid(m-1,n,6,2)|lamid(5,2)|[2,2,0; 5,1,0]#
   rinid(m,n-1,6,2)|[2,2,0; 2,4,0]#
   rinid(m-1,n,6,2)|[5,1,0; 5,3,0];

W=rinid(m,n,6,2)|(lamid(2,2)|[1,1,1; 2,2,0]#
   lam(1,2)|[0,2,1; 2,2,0])#

   rinid(m-1,n+1,6,2)|(lamid(5,1)|

   [4,0,1; 5,1,0]#lam(1,5)|[3,1,1; 5,1,0]);
Grid=bt(L1/(6*m-2),L2/((n+1)*2),D)|

   (T#B#W);

use &,vm(2),vt(2),vh(L1/2,-L2/2,4*L2,

   L1/2,L2/2,0, L1/2,L2/2,1);

clear;   draw Grid;
برنامهء فرميينی برای تاشه های شبيه به شکل 52
(*) Formian scheme for configurations

   similar to Fig 52:  Double layer grid with

   hexagon on larger hexagon pattern.

   See Reference 1 (*)

L1=24;   (*) Length in U1 direction (*)

L2=21;   (*) Length in U2 direction (*)

D=2;      (*) Depth of grid (*)

m=5;      (*) Frequency in U1 direction (*)

n=6;       (*) Frequency in U2 direction,

                                      n must be even (*)

T=rinid(m,n,6,2)|lamid(2,2)|[1,1,1; 0,2,1]#

   rin(1,m-1,6)|lamid(5,n+1)|[3,1,1; 4,0,1]#

   rinid(m,n+1,6,2)|[1,1,1; 3,1,1]#

   rinid(m-1,n+2,6,2)|[4,0,1; 6,0,1];

RN=rinid(m-1,n/2,6,4)|[5,3,0];

B=lux(RN)|(rinid(m-1,n,6,2)|lamid(5,2)|
   [2,2,0; 5,1,0]#rinid(m,n-1,6,2)|[2,2,0;
   2,4,0]#rinid(m-1,n,6,2)|[5,1,0; 5,3,0]);

W=lux(RN)|(rinid(m,n,6,2)|(lamid(2,2)|
   [1,1,1; 2,2,0]#lam(1,2)|[0,2,1; 2,2,0])#

   rinid(m-1,n+1,6,2)|(lamid(5,1)|

   [4,0,1; 5,1,0]#lam(1,5)|[3,1,1; 5,1,0]));

Grid=bt(L1/(6*m-2),L2/((n+1)*2),D)|

   (T#B#W);

use &,vm(2),vt(2),vh(L1/2,-L2/2,4*L2,

   L1/2,L2/2,0, L1/2,L2/2,1);

clear;   draw Grid;
برنامهء فرميينی برای تاشه های شبيه به شکل 53
(*) Formian scheme for configurations

   similar to Fig 53:  Binate trussed double

   layer grid. See Reference 1 (*)

L1=16;   (*) Length in U1 direction (*)

L2=16;   (*) Length in U2 direction (*)

D=1.5;    (*) Depth of grid (*)

m=5;       (*) Frequency in U1 direction (*)

n=5;        (*) Frequency in U2 direction (*)

T=rinid(m,n+1,2,2)|[0,0,1; 2,0,1]#

   rinid(m+1,n,2,2)|[0,0,1; 0,2,1];

B=rinid(m,n+1,2,2)|[0,0,0; 2,0,0]#

   rinid(m+1,n,2,2)|[0,0,0; 0,2,0];

E=lamis(1,0.5)|[0,0,0; 1,0,0.5]#

   lam(1,1)|[0,0,0; 0,0,1];

W=pex|(rinid(m,n+1,2,2)|E#

   rinid(m+1,n,2,2)|verad(0,0)|E);

Grid=bt(L1/(2*m),L2/(2*n),D)|(T#B#W);

use &,vm(2),vt(2),vh(L1/2,-L2/3,2*L2,

   L1/2,L2/2,0, L1/2,L2/2,1);

clear;   draw Grid;
برنامهء فرميينی برای تاشه های شبيه به شکل 54
(*) Formian scheme for configurations

   similar to Fig 54:  Obnate trussed

   double layer grid. See Reference 1 (*)

L1=16;   (*) Length in U1 direction (*)

L2=16;   (*) Length in U2 direction (*)

D=1.5;   (*) Depth of grid (*)

m=4;      (*) Frequency in U1 direction (*)

n=4;       (*) Frequency in U2 direction (*)

T=rinid(m,n,2,2)|lamid(1,1)|[0,0,1; 1,1,1]#

   rinid(m,2,2,2*n)|[0,0,1; 2,0,1]#

   rinid(2,n,2*m,2)|[0,0,1; 0,2,1];

B=rinid(m,n,2,2)|lamid(1,1)|[0,0,0; 1,1,0]#

   rinid(m,2,2,2*n)|[0,0,0; 2,0,0]#

   rinid(2,n,2*m,2)|[0,0,0; 0,2,0];

E=lam(3,0.5)|{[0,0,0; 0.5,0.5,0.5],

   [0.5,0.5,0.5; 1,1,0]}#

   {[0,0,1; 0,0,0], [1,1,1; 1,1,0]};

F=lamis(1,0.5)|[0,0,0; 1,0,0.5]#

   lam(1,1)|[0,0,0; 0,0,1];

W=pex|(rinid(m,n,2,2)|lamid(1,1)|E#

   rinid(m,2,2,2*n)|F#rinid(2,n,2*m,2)|

   verad(0,0)|F);

Grid=bt(L1/(2*m),L2/(2*n),D)|(T#B#W);

use &,vm(2),vt(2),vh(L1/2,-L2,2.5*L2,

   L1/2,L2/2,0, L1/2,L2/2,1);

clear;   draw Grid;
برنامهء فرميينی برای تاشه های شبيه به شکل 55
(*) Formian scheme for configurations

   similar to Fig 55:  Double layer grid with

   ternate on ternate pattern and with a

   triangular shape in plan.

   See Section 1.7 of Reference 1. (*)

L=20;   (*) Length of each side of grid (*)

D=2;    (*) Depth of grid (*)

m=7;    (*) Number of tetrahedral units

                   along each side of the grid (*)

T=genid(m,m,2,3,1,-1)|([0,0,1; 2,0,1]#
   lam(1,1)|[0,0,1; 1,3,1]);
B=genid(m-1,m-1,2,3,1,-1)|([1,1,0; 3,1,0]#
   lam(1,2)|[1,1,0; 2,4,0]);

W=genid(m,m,2,3,1,-1)|([1,3,1; 1,1,0]#
   lam(1,1)|[0,0,1; 1,1,0]);

Grid=bt(L/(2*m),L*sqrt|3/(6*m),D)|
   (T#B#W);

use &,vm(2),vt(2),vh(L/2,-L/2,6*L,

   L/2,L/2,0, L/2,L/2,1);

clear;   draw Grid;
برنامهء فرميينی برای تاشه های شبيه به 
شکلهای 56 و 57
(*) Formian scheme for configurations

   similar to Fig 56 (m=12, n=3) and

   Fig 57 (m=12, n=4):  Double layer grid

   with ternate on ternate pattern and

   a triangular plan with curtailed corners.

   See Section 1.7 of Reference 1. (*)

L=40;    (*) Length of each side of the

                   complete triangular plan (*)

D=2;     (*) Depth of grid (*)

m=12;   (*) Number of tetrahedral units

   along each side of the complete

   triangular plan (*)

n=3;    (*) Extent of the corner

                      curtailment of the grid (*)

RN=lam(1,m)|genid(n,n,2,3,1,-1)|
   {[0,0,1],[1,1,0]}#genid(1,n,2,-3,-1,1)|

   {[m,3*m,1], [m,3*m-2,0]};

T=lux(RN)|genid(m,m,2,3,1,-1)|
   ([0,0,1; 2,0,1]#lam(1,1)|[0,0,1; 1,3,1]);

B=lux(RN)|genid(m-1,m-1,2,3,1,-1)|
   ([1,1,0; 3,1,0]#lam(1,2)|[1,1,0; 2,4,0]);

W=lux(RN)|genid(m,m,2,3,1,-1)|
   ([1,3,1; 1,1,0]#lam(1,1)|[0,0,1; 1,1,0]);

Grid=bt(L/(2*m),L*sqrt|3/(6*m),D)|

   (T#B#W);

use &,vm(2),vt(2),vh(L/2,-L/2,6*L,

   L/2,L/2,0, L/2,L/2,1);

clear;   draw Grid;
برنامهء فرميينی برای تاشه های شبيه 
به شکلهای 58، 59 و 60
(*) Formian scheme for configurations

   similar to Fig 58 (m=5, n=5), Fig 59
   (m=4, n=6) and Fig 60 (m=4, n=8):

   Sectorate double layer grid with ternate

   on ternate pattern. See Section 2.2 of

   Reference 2. (*)

L=20;      (*) Length of each side (*)

D=2.15;  (*) Depth of grid (*)

m=5;       (*) Frequency of top layer (*)

n=5;        (*) Number of sides (*)

S=L/(2*tan|(180/n));

ET=genid(1,m,2,3,-1,1)|{[0,0,0; -1,3,0],

   [0,0,0; 1,3,0], [-1,3,0; 1,3,0]};

EB=genid(1,m-1,2,3,-1,1)|{[0,2,-1; -1,5,-1],

   [0,2,-1; 1,5,-1], [-1,5,-1; 1,5,-1]};

EW=genid(1,m,2,3,-1,1)|{[0,0,0; 0,2,-1],

   [-1,3,0; 0,2,-1], [1,3,0; 0,2,-1]};

E=bt(L/(2*m),S/(3*m),D)|(ET#EB#EW);

Qi=[0,2,-1];   Ri=bt(L/(2*m),S/(3*m),D)|Qi;

Rj=verad(0,0,360/n)|Ri;   K1=tig|(Ri#Rj);

K2=rinad(0,0,-L/(2*m),S/m,m)|K1;

Grid=pex|rosad(0,0,n,360/n)|(E#K2);

use &,vm(2);   clear;   draw Grid;
برنامهء فرميينی برای تاشه های شبيه 

به شکلهای 61، 62 و 63
(*) Formian scheme for configurations

   similar to Fig 61 (n=5), Fig 62 (n=6)

   and Fig 63 (n=7):  Sectorate double layer

   grid with ternate/hexagonal on ternate/

   hexagonal pattern. See Reference 1. (*)

L=20;  (*) Length of each side of the grid (*)

D=2;    (*) Depth of grid (*)

m=8;    (*) Frequency of the original ternate
                  pattern of the top layer, m must

                         be a multiple of 3 minus 1 (*)

n=5;     (*) Number of sides (*)

q=ric|((m+1)/3);

E=[-2,2,1; 2,2,1]#lam(1,0)|{[-2,2,1; 0,4,1],

   [-4,4,1; 0,4,1]};

T=genid(1,q,12,6,-6,1)|E#

   genid(1,q-1,12,6,-6,1)|(lamid(0,6)|

   lam(1,-2)|[-4,4,1; -2,6,1]#ref(2,6)|E);

B=genid(1,q-1,12,6,-6,1)|lamid(0,6)|[-3,3,0;

   0,6,0]#genid(1,q,6,6,-6,2)|[-3,3,0; 3,3,0];

W=(genid(1,q,12,6,-6,1)|lam(1,0)|

   {[0,4,1; -3,3,0], [-2,2,1; -3,3,0],

   [-4,4,1; -3,3,0]}#genid(1,q-1,12,6,-6,1)|

   (lam(1,0)|{[-2,6,1; 0,6,0], [0,8,1; -3,9,0]}#

   lam(2,6)|[0,4,1; 0,6,0]));

Grid=pex|rosad(0,0,n,360/n)|bt(L/(4*m),

   L/(4*m*tan|(180/n)),D)|(T#B#W);

use &,vm(2);   clear;   draw Grid;

برنامهء فرميينی برای تاشه های شبيه به 
شکلهای 64 و 65
(*) Formian scheme for configurations

   similar to Fig 64 (p=6, q=6) and Fig 65
   (p=4, q=2):  Compound double layer

   grid with binate on binate pattern

   (obtained by overlapping two grids).

   See Section 1.7 of Reference 1. (*)

L1=16;    (*) Length of the base grid

                                     in U1 direction (*)

L2=16;    (*) Length of the base grid

                                     in U2 direction (*)

D=2;       (*) Depth of grid (*)

m=5;       (*) Frequency of the base grid

                                     in U1 direction (*)

n=5;        (*) Frequency of the base grid

                                     in U2 direction (*)

p=6;        (*) Shift in U1 direction,

                                     p must be even (*)

q=6;        (*) Shift in U2 direction,

                                     q must be even (*)

T=rinid(m,n+1,2,2)|[0,0,1; 2,0,1]#

   rinid(m+1,n,2,2)|[0,0,1; 0,2,1];

B=rinid(m-1,n,2,2)|[1,1,0; 3,1,0]#

   rinid(m,n-1,2,2)|[1,1,0; 1,3,0];

W=rinid(m,n,2,2)|lamid(1,1)|[0,0,1; 1,1,0];

G=T#B#W;

G=pex|(G#tranid(p,q)|G);

Grid=bt(L1/(2*m),L2/(2*n),D)|G;

use &,vm(2),vt(2),vh(L1/2,-L2/3,3*L2,

   L1/2,L2/2,0, L1/2,L2/2,1);

clear;   draw Grid;
برنامهء فرميينی برای تاشه های شبيه به شکل 66
(*) Formian scheme for configurations

   similar to Fig 66:  Barrel vault with binate

   pattern. See Section 1.8 of Reference 1. (*)

S=28;  (*) Span (*)

H=8;   (*) Rise (*)

L=37;  (*) Length (*)
m=9;   (*) Frequency in U2 direction (*)

n=12;  (*) Frequency in U3 direction (*)

A=2*atan|(2*H/S);   (*) Sweep angle (*)

R=S/(2*sin|A);          (*) Radius (*)

E=rinit(m,n+1,1,1)|[1,0,0; 1,1,0]#

   rinit(m+1,n,1,1)|[1,0,0; 1,0,1];

F=bc(R,2*A/m,L/n)|E;

Vault=verad(0,0,90-A)|F;

use &,vm(2),vt(2),

   vh(0,2.5*R,-R, 0,0,R, 0,1,R);

clear;   draw Vault;
برنامهء فرميينی برای تاشه های شبيه به شکل 67
(*) Formian scheme for configurations

   similar to Fig 67:  Barrel vault with lamella

   (obnate) pattern. See Section 1.8 of
   Reference 1. (*)

S=30;   (*) Span (*)

H=8;    (*) Rise (*)

L=45;   (*) Length (*)

m=7;    (*) Frequency in U2 direction (*)

n=8;     (*) Frequency in U3 direction (*)

A=2*atan|(2*H/S);   (*) Sweep angle (*)

R=S/(2*sin|A);          (*) Radius (*)

E=rinit(m,n,2,2)|lamit(1,1)|[1,0,0; 1,1,1]#

   rinit(m,2,2,2*n)|[1,0,0; 1,2,0]#

   rinit(2,n,2*m,2)|[1,0,0; 1,0,2];

F=bc(R,A/m,L/(2*n))|E;

Vault=verad(0,0,90-A)|F;

use &,vm(2),vt(2),

   vh(0,3*R,-2*R, 0,0,R, 0,1,R);

clear;   draw Vault;
برنامهء فرميينی برای تاشه های شبيه به شکل 68
(*) Formian scheme for configurations

   similar to Fig 68:  Lamella barrel vault

   with additional ribs.

   See Section 1.8 of Reference 1. (*)

S=30;   (*) Span (*)

H=8;    (*) Rise (*)

L=45;   (*) Length (*)

m=7;    (*) Frequency in U2 direction (*)

n=8;     (*) Frequency in U3 direction (*)

A=2*atan|(2*H/S);   (*) Sweep angle (*)

R=S/(2*sin|A);          (*) Radius (*)

E=rinit(m,n,2,2)|lamit(1,1)|[1,0,0; 1,1,1]#

  rinit(m,n+1,2,2)|[1,0,0; 1,2,0]#

   rinit(2,n,2*m,2)|[1,0,0; 1,0,2];

F=bc(R,A/m,L/(2*n))|E;

Vault=verad(0,0,90-A)|F;

use &,vm(2),vt(2),

   vh(0,3*R,-2*R, 0,0,R, 0,1,R);

clear;   draw Vault;
برنامهء فرميينی برای تاشه های شبيه به شکل 69
(*) Formian scheme for configurations

   similar to Fig 69:  Binate barrel vault

   with additional bracings.

   See Section 1.8 of Reference 1. (*)

S=28;   (*) Span (*)

H=8;    (*) Rise (*)

L=32;   (*) Length (*)

m=8;    (*) Frequency 

          in U2 direction, m must be even (*)

n=10;   (*) Frequency

          in U3 direction, n must be even (*)

A=2*atan|(2*H/S);   (*) Sweep angle (*)

R=S/(2*sin|A);          (*) Radius (*)

mm=ric|(m/2);   nn=ric|(n/2);

E=rinit(m,n+1,1,1)|[1,0,0; 1,1,0]#

   rinit(m+1,n,1,1)|[1,0,0; 1,0,1]#

   lamit(mm,nn)|rinit(mm,nn,1,1)|

   [1,0,1; 1,1,0];

F=bc(R,2*A/m,L/n)|E;

Vault=verad(0,0,90-A)|F;

use &,vm(2),vt(2),

   vh(0,3*R,-R, 0,0,R, 0,1,R);

clear;   draw Vault;
برنامهء فرميينی برای تاشه های شبيه به شکل 70
(*) Formian scheme for configurations

   similar to Fig 70:  Binate barrel vault

   with additional bracings.

   See Section 1.8 of Reference 1. (*)

S=28;   (*) Span (*)

H=8;    (*) Rise (*)

L=32;   (*) Length (*)  

m=10;  (*) Frequency 

          in U2 direction, m must be even (*)

n=10;   (*) Frequency

          in U3 direction, n must be even (*)

A=2*atan|(2*H/S);   (*) Sweep angle (*)

R=S/(2*sin|A);          (*) Radius (*)

mm=ric|(m/2);   nn=ric|(n/2);

E=lux(rin(3,nn,2)|[1,mm,1])|

   (rinit(m,n+1,1,1)|[1,0,0; 1,1,0]#

   rinit(m+1,n,1,1)|[1,0,0; 1,0,1]#

   lamit(mm,nn)|rinit(mm,ric|(nn/2),1,2)|

   lam(3,1)|[1,0,1; 1,1,0]);

F=bc(R,2*A/m,L/n)|E;

Vault=verad(0,0,90-A)|F;

use &,vm(2),vt(2),

   vh(0,3*R,-R, 0,0,R, 0,1,R);

clear;   draw Vault;
برنامهء فرميينی برای تاشه های شبيه به شکل 71
(*) Formian scheme for configurations

   similar to Fig 71:  Double layer barrel 

   vault with binate on binate pattern.

   See Section 1.8 of Reference 1. (*)

S=20;    (*) Span (of top layer) (*)

H=6;     (*) Rise (of top layer) (*)

D=1.5;  (*) Depth of the vault (*)

L=24;    (*) Length of the vault (*)

m=6;     (*) Frequency in U2 direction (*)

n=6;      (*) Frequency in U3 direction (*)

A=2*atan|(2*H/S);  (*) Sweep angle (*)

Rt=S/(2*sin|A);       (*) Top radius (*)

Rb=Rt-D;                (*) Bottom radius (*)

T=rinit(m,n+1,2,2)|[Rt,0,0; Rt,2,0]#

   rinit(m+1,n,2,2)|[Rt,0,0; Rt,0,2];

B=rinit(m-1,n,2,2)|[Rb,1,1; Rb,3,1]#

   rinit(m,n-1,2,2)|[Rb,1,1; Rb,1,3];

W=rinit(m,n,2,2)|lamit(1,1)|

   [Rt,0,0; Rb,1,1];

E=bc(1,A/m,L/(2*n))|(T#B#W);

Vault=verad(0,0,90-A)|E;

use &,vm(2),vt(2),

   vh(0,3*Rt,-Rt, 0,0,Rt, 0,1,Rt);

clear;   draw Vault;
برنامهء فرميينی برای تاشه های شبيه به شکل 72
(*) Formian scheme for configurations

   similar to Fig 72:  Compound barrel vault.

   See Section 1.8 of Reference 1. (*)

S=20;   (*) Span (*)

H=7;    (*) Rise (*)

L=36;   (*) Length (*)

m=6;    (*) Frequency in U2 direction (*)

n=8;     (*) Frequency in U3 direction (*)

A=2*atan|(2*H/S);   (*) Sweep angle (*)

R=S/(2*sin|A);          (*) Radius (*)

E=rinit(m,n,2,2)|lamit(1,1)|[1,0,0; 1,1,1]#

  rinit(m,n+1,2,2)|[1,0,0; 1,2,0]#

   rinit(2,n,2*m,2)|[1,0,0; 1,0,2];

F=bc(R,A/m,L/(2*n))|E;

Vault=rin(1,2,S)|verad(0,0,90-A)|F;

use &,vm(2),vt(2),

   vh(S/2,6*R,-3*R, S/2,0,R, S/2,1,R);

clear;   draw Vault;
برنامهء فرميينی برای تاشه های شبيه به شکل 73
(*) Formian scheme for configurations

   similar to Fig 73:  Compound barrel vault.

   See Section 1.8 of Reference 1. (*)

S=16;   (*) Span of each barrel vault (*)

H=4;    (*) Rise (*)

L=40;   (*) Length (*)  

m=6;    (*) Frequency 

          in U2 direction, m must be even (*)

n=10;   (*) Frequency

          in U3 direction, n must be even (*)

A=2*atan|(2*H/S);   (*) Sweep angle (*)

R=S/(2*sin|A);          (*) Radius (*)

mm=ric|(m/2);   nn=ric|(n/2);

E=lux(rin(3,nn,2)|[1,mm,1])|

   (rinit(m,n+1,1,1)|[1,0,0; 1,1,0]#

   rinit(m+1,n,1,1)|[1,0,0; 1,0,1]#

   lamit(mm,nn)|rinit(mm,ric|(nn/2),1,2)|

   lam(3,1)|[1,0,1; 1,1,0]);

F=bc(R,2*A/m,L/n)|E;

Vault=rin(1,3,S)|verad(0,0,90-A)|F;

use &,vm(2),vt(2),

   vh(S,8*R,-4*R, S,0,R, S,1,R);

clear;   draw Vault;
برنامهء فرميينی برای تاشه های شبيه به شکل 74
(*) Formian scheme for configurations

   similar to Fig 74:  Sidebent annular barrel

   vault with binate pattern.

   See Section 3.4 of Reference 3. (*)

S=24;   (*) Span of barrel vault (*)

H=6;    (*) Rise of barrel vault (*)

m=20;  (*) Frequency in U2 direction (*)

n=10;    (*) Frequency in U4 direction (*)

R=60;   (*) Radius of annulus (*)

T=60;   (*) Central angle

                    containing the barrel vault (*)

A=2*atan|(2*H/S);    (*) Sweep angle (*)

Q=S/(2*sin|A);          (*) Radius of

                      cross-section of annulus (*)

E=rinic(2,4,m,n+1,1,1)|[1,0,1,-n/2;

   1,1,1,-n/2]#rinic(2,4,m+1,n,1,1)|

   [1,0,1,-n/2; 1,0,1,1-n/2];

Vault=dep(4)|ba(R,T/m,Q,2*A/n)|E;

use &,vm(2),vt(2),

   vh(R/2,2*R,1.5*R, 0,0,0, 0,0,1);

clear;   draw Vault;
برنامهء فرميينی برای تاشه های شبيه به شکل 75
(*) Formian scheme for configurations

   similar to Fig 75:  Sidebent annular barrel

   vault with lamella (obnate) pattern.

   See Section 3.4 of Reference 3. (*)

S=28;   (*) Span of barrel vault (*)

H=8;    (*) Rise of barrel vault (*)

m=16;  (*) Frequency in U2 direction (*)

n=8;    (*) Frequency in U4 direction (*)

R=70;   (*) Radius of annulus (*)

T=50;   (*) Central angle

                    containing the barrel vault (*)

A=2*atan|(2*H/S);    (*) Sweep angle (*)

Q=S/(2*sin|A);          (*) Radius of

                      cross-section of annulus (*)

E=rinic(2,4,m,n,2,2)|lam(4,1-n)|lam(2,1)|

   [1,0,1,-n; 1,1,1,1-n]#

   rinic(2,4,m,2,2,2*n)|

   [1,0,1,-n; 1,2,1,-n]#

   rinic(2,4,2,n,2*m,2)|

   [1,0,1,-n; 1,0,1,2-n];

Vault=dep(4)|ba(R,T/(2*m),Q,A/n)|E;

use &,vm(2),vt(2),

   vh(R,2*R,2*R, 0,0,0, 0,0,1);

clear;   draw Vault;
برنامهء فرميينی برای تاشه های شبيه به شکل 76
(*) Formian scheme for configurations

   similar to Fig 76:  Sidebent horseshoe

   annular barrel vault with binate pattern.

   See Section 3.4 of Reference 3. (*)

S=60;    (*) Span of barrel vault (*)

H=15;   (*) Rise of barrel vault (*)

m=48;   (*) Frequency in U2 direction (*)

n=7;     (*) Frequency in U4 direction (*)

R=90;   (*) Radius of annulus (*)

T=270; (*) Central angle

                    containing the barrel vault (*)

A=2*atan|(2*H/S);    (*) Sweep angle (*)

Q=S/(2*sin|A);          (*) Radius of

                      cross-section of annulus (*)

E=rinic(2,4,m,n+1,1,1)|

   [1,0,1,-n/2; 1,1,1,-n/2]#

   rinic(2,4,m+1,n,1,1)|

   [1,0,1,-n/2; 1,0,1,1-n/2];

Vault=dep(4)|ba(R,T/m,Q,2*A/n)|E;

use &,vm(2),vt(2),

   vh(2*R,2*R,4*R, 0,0,0, 0,0,1);

clear;   draw Vault;
برنامهء فرميينی برای تاشه های شبيه به شکل 77
(*) Formian scheme for configurations

   similar to Fig 77:  Sidebent annular ring

   barrel vault with lamella (obnate) pattern.

   See Section 3.4 of Reference 3. (*)

S=50;   (*) Span of barrel vault (*)

H=15;   (*) Rise of barrel vault (*)

m=30;   (*) Frequency in U2 direction (*)

n=5;      (*) Frequency in U4 direction (*)

R=80;   (*) Radius of annulus (*)

T=360; (*) Central angle

                    containing the barrel vault (*)

A=2*atan|(2*H/S);    (*) Sweep angle (*)

Q=S/(2*sin|A);          (*) Radius of

                      cross-section of annulus (*)

E=rinic(2,4,m,n,2,2)|lam(4,1-n)|lam(2,1)|

   [1,0,1,-n; 1,1,1,1-n]#

   rinic(2,4,m,2,2,2*n)|

   [1,0,1,-n; 1,2,1,-n];

Vault=dep(4)|ba(R,T/(2*m),Q,A/n)|E;

use &,vm(2),vt(2),

   vh(R,R,3*R, 0,0,0, 0,0,1);

clear;   draw Vault;
برنامهء فرميينی برای تاشه های شبيه به شکل 78
(*) Formian scheme for configurations

   similar to Fig 78:  Outbent annular

   barrel vault with lamella (obnate) pattern

   and additional ribs.

   See Section 3.4 of Reference 3. (*)

S=48;   (*) Span of barrel vault (*)

H=10;   (*) Rise of barrel vault (*)

m=12;   (*) Frequency in U2 direction (*)

n=5;      (*) Frequency in U4 direction (*)

R=50;   (*) Radius of annulus (*)

T=90;   (*) Central angle

                    containing the barrel vault (*)

A=2*atan|(2*H/S);    (*) Sweep angle (*)

Q=S/(2*sin|A);          (*) Radius of

                      cross-section of annulus (*)

E=rinic(2,4,m,n,2,2)|lam(4,1-n)|lam(2,1)|

   [1,0,1,-n; 1,1,1,1-n]#

   rinic(2,4,m,2,2,2*n)|

   [1,0,1,-n; 1,2,1,-n]#

   rinic(2,4,m+1,n,2,2)|

   [1,0,1,-n; 1,0,1,2-n];

E=tran(4,n*90/A)|E;

Vault=dep(4)|ba(R,T/(2*m),Q,A/n)|E;

use &,vm(2),vt(2),

   vh(5*R,10*R,5*R, 0,0,0, 1,1,0);

clear;   draw Vault;

برنامهء فرميينی برای تاشه های شبيه به شکل 79
(*) Formian scheme for configurations

   similar to Fig 79:  Compound outbent

   annular barrel vault with lamella pattern.

   See Section 3.5 of Reference 3. (*)

S=48;     (*) Span of barrel vault (*)

H=12;    (*) Rise of barrel vault (*)

m=10;    (*) Frequency in U2 direction (*)

n=5;       (*) Frequency in U4 direction (*)

R=50;     (*) Radius of annulus (*)

T=90;     (*) Central angle

                    containing the barrel vault (*)

A=2*atan|(2*H/S);    (*) Sweep angle (*)

Q=S/(2*sin|A);    (*) Radius of

                      cross-section of annulus (*)

E=rinic(2,4,m,n,2,2)|lam(4,1-n)|lam(2,1)|

   [1,0,1,-n; 1,1,1,1-n]#

   rinic(2,4,m,2,2,2*n)|[1,0,1,-n; 1,2,1,-n]#

   rinic(2,4,2,n,2*m,2)|[1,0,1,-n; 1,0,1,2-n];

E=tran(4,n*90/A)|E;

B=dep(4)|ba(R,T/(2*m),Q,A/n)|E;

Vault=rin(3,3,S)|B;

use &,vm(2),vt(2),

   vh(5*R,10*R,5*R, 0,0,0, 1,1,0);

clear;   draw Vault;
برنامهء فرميينی برای تاشه های شبيه به شکل 80
(*) Formian scheme for configurations

   similar to Fig 80:  Inbent annular barrel

   vault with binate pattern.

   See Section 3.4 of Reference 3. (*)

S=40;     (*) Span of barrel vault (*)

H=8;      (*) Rise of barrel vault (*)

m=16;    (*) Frequency in U2 direction (*)

n=7;       (*) Frequency in U4 direction (*)

R=100;  (*) Radius of annulus (*)

T=90;    (*) Central angle

                    containing the barrel vault (*)

A=2*atan|(2*H/S);    (*) Sweep angle (*)

Q=S/(2*sin|A);          (*) Radius of

                      cross-section of annulus (*)

E=rinic(2,4,m,n+1,1,1)|

   [1,0,1,-n/2; 1,1,1,-n/2]#

   rinic(2,4,m+1,n,1,1)|

   [1,0,1,-n/2; 1,0,1,1-n/2];

F=tran(4,-n*90/(2*A))|E;

Vault=dep(4)|ba(R,T/m,Q,2*A/n)|F;

use &,vm(2),vt(2),

   vh(1.5*R,1.5*R,R, 0,0,0, 0,0,1);

clear;   draw Vault;
برنامهء فرميينی برای تاشه های شبيه به شکل 81
(*) Formian scheme for configurations

   similar to Fig 81:  Compound inbent

   annular barrel vault with binate pattern.

   See Section 3.5 of Reference 3. (*)

S=40;     (*) Span of barrel vault (*)

H=8;      (*) Rise of barrel vault (*)

m=16;    (*) Frequency in U2 direction (*)

n=7;       (*) Frequency in U4 direction (*)

R=100;  (*) Radius of annulus (*)

T=90;    (*) Central angle

                    containing the barrel vault (*)

A=2*atan|(2*H/S);    (*) Sweep angle (*)

Q=S/(2*sin|A);          (*) Radius of

                      cross-section of annulus (*)

E=rinic(2,4,m,n+1,1,1)|

   [1,0,1,-n/2; 1,1,1,-n/2]#

   rinic(2,4,m+1,n,1,1)|

   [1,0,1,-n/2; 1,0,1,1-n/2];

F=tran(4,-n*90/(2*A))|E;

G=dep(4)|ba(R,T/m,Q,2*A/n)|F;

Vault=rin(3,3,S)|G;

use &,vm(2),vt(2),

   vh(6*R,6*R,8*R, 0,0,0, 0,0,1);

clear;   draw Vault;
برنامهء فرميينی برای تاشه های شبيه به شکل 82
(*) Formian scheme for configurations

   similar to Fig 82:  Double layer inbent 

   annular barrel vault with binate on

   binate pattern. See Section 3.4 of 

   Reference 3. (*)

S=45;     (*) Span of top layer (*)

H=6;      (*) Rise of top layer (*)

D=2;      (*) Depth of barrel vault (*)

m=8;  (*) Top layer frequency along U2 (*)

n=5;   (*) Top layer frequency along U4 (*)

R=100;  (*) Radius of annulus (*)

TT=70;    (*) Central angle containing the

                       top layer of barrel vault (*)

A=2*atan|(2*H/S);    (*) Sweep angle (*)

Qt=S/(2*sin|A);          (*) Radius of top

              layer cross-section of annulus (*)

Qb=Qt-D;

T=rinic(2,4,m,n+1,2,2)|

   [1,0,Qt,-n; 1,2,Qt,-n]#

   rinic(2,4,m+1,n,2,2)|

   [1,0,Qt,-n; 1,0,Qt,2-n];

B=rinic(2,4,m-1,n,2,2)|

   [1,1,Qb,1-n; 1,3,Qb,1-n]#

   rinic(2,4,m,n-1,2,2)|

   [1,1,Qb,1-n; 1,1,Qb,3-n];

W=rinic(2,4,m,n,2,2)|lamic(2,4,1,1-n)|

   [1,0,Qt,-n; 1,1,Qb,1-n];

E=tran(4,-2*n*90/(2*A))|(T#B#W);

Vault=dep(4)|ba(R,TT/(2*m),1,A/n)|E;

use &,vm(2),vt(2),

   vh(2*R,1.5*R,R, 0,0,0, 0,0,1);

clear;   draw Vault;

برنامهء فرميينی برای تاشه های شبيه 
به شکلهای 83 و 84
(*) Formian scheme for configurations

   similar to Fig 83 (R1=16, R2=10) and
   Fig 84 (R1=10, R2=15):  Parabolic barrel

   vault. See Section 2.3 of  Reference 2. (*)

R1=16;   (*) Radius at the ends (*)

R2=10;   (*) Radius at the middle (*)

L=36;     (*) Length of barrel vault (*)

P=60;     (*) Sweep angle (*)

m=12;    (*) Frequency in U2 direction (*)

n=20;     (*) Frequency in U3 direction (*)

a=4*(R1-R2)/L^2;   b=-a*L;

c=R1;   d=L/n;

E=rin(2,m,1)|lib(i=0,n)|

   [a*(d*i)^2+b*d*i+c,0,i;

   a*(d*i)^2+b*d*i+c,1,i]#

   rin(2,m+1,1)|lib(i=0,n-1)|

   [a*(d*i)^2+b*d*i+c,0,i;

   a*(d*(i+1))^2+b*d*(i+1)+c,0,i+1];

Vault=verad(0,0,90-P)|bc(1,2*P/m,d)|E;

use &, vm(2),vt(2),

   vh(L,3*L,-L, 0,0,L, 0,1,L);

clear;   draw Vault;

برنامهء فرميينی برای تاشه های شبيه 
به شکلهای 85 و 86
(*) Formian scheme for configurations

   similar to Fig 85 (R1=16, R2=12) and
   Fig 86 (R1=12, R2=16):

   Hyperboloidal barrel vault.

   See Section 3.3.2 of Reference 3. (*)

R1=16;   (*) Radius at the ends (*)

R2=11;   (*) Radius at the middle (*)

L=36;     (*) Length of barrel vault (*)

A=60;     (*) Sweep angle (*)

m=4;  (*) Circumferential semifrequency (*)

n=6;   (*) Longitudinal semifrequency (*)

P=(R1^2-R2^2)/(L/2)^2;    (*) Pitch (*)

E=pex|lamit(0,0)|rinit(m,n,1,1)|

   (rosat(0.5,0.5)|[1,0,0; 1,1,0]#

   [1,0,1; 1,1,0]);

F=bh(R2,A/m,L/(2*n),P)|E;

Vault=verad(0,0)|F;

use &, vm(2),vt(2),

   vh(L,3*L,-L, 0,0,L, 0,1,L);

clear;   draw Vault;

برنامهء فرميينی برای تاشه های شبيه 

به شکل 87 و 88
(*) Formian scheme for configurations

   similar to Fig 87 (C=3, parabolic) and

   Fig 88 (C=4, Wedge-shaped): Pellevated

   honeycomb grid.

   See References 4, 5 and 6. (*)

S=20;     (*) Span (*)

L=24;     (*) Length (*)

H=7;      (*) Rise (*)

m=6;      (*) Circuferential frequency,

    number of hexagons in the second row (*)

n=7;       (*) Longitudinal frequency,

      number of hexagons along the sides (*)

C=3;      (*) Pellevation code (*)

E=rinid(m,n,6,2)|lamid(2,2)|[1,1,0; 0,2,0]#

   rinid(m,n+1,6,2)|[1,1,0; 3,1,0]#

   rin(1,m-1,6)|lamid(5,n+1)|[4,0,0; 3,1,0]#

   rinid(m-1,n+2,6,2)|[4,0,0; 6,0,0];

Grid=bb(S/(6*m-2),L/(2*(n+1)))|E;

Vault=bapel(C,S/2,0,S/2,1,S,H)|Grid;

use &,vm(2),vt(2),vh(L/2,-2*L,3*L,

   S/2,L/2,0, S/2,L/2,1);

clear;   draw Vault;
برنامهء فرميينی برای تاشه های شبيه 

به شکل 89 و 90
(*) Formian scheme for configurations

   similar to Fig 89 (C=3, parabolic) and

   Fig 90 (C=4, Wedge-shaped): Pellevated

   trussed lamella double layer barrel vault.

   See References 4, 5 and 6. (*)

S=20;     (*) Span (*)

L=20;     (*) Length (*)

D=2;      (*) Depth of grid (*)

H=6;      (*) Rise (*)

m=4;      (*) Circuferential frequency (*)

n=4;       (*) Longitudinal frequency (*)

C=3;      (*) Pellevation code (*)

T=rinid(m,n,2,2)|lamid(1,1)|[0,0,1; 1,1,1]#

   rinid(m,2,2,2*n)|[0,0,1; 2,0,1]#

   rinid(2,n,2*m,2)|[0,0,1; 0,2,1];

B=rinid(m,n,2,2)|lamid(1,1)|[0,0,0; 1,1,0]#

   rinid(m,2,2,2*n)|[0,0,0; 2,0,0]#

   rinid(2,n,2*m,2)|[0,0,0; 0,2,0];

E=lam(3,0.5)|{[0,0,0; 0.5,0.5,0.5],

   [0.5,0.5,0.5; 1,1,0]}#

   {[0,0,1; 0,0,0], [1,1,1; 1,1,0]};

F=lamis(1,0.5)|[0,0,0; 1,0,0.5]#

   lam(1,1)|[0,0,0; 0,0,1];

W=pex|(rinid(m,n,2,2)|lamid(1,1)|E#

   rinid(m,2,2,2*n)|F#rinid(2,n,2*m,2)|

   verad(0,0)|F);

Grid=bt(S/(2*m),L/(2*n),D)|(T#B#W);

Vault=bapel(C,S/2,0,S/2,1,S,H)|Grid;

use &,vm(2),vt(2),vh(L/2,-2*L,4*L,

   S/2,L/2,0, S/2,L/2,1);

clear;   draw Vault;
برنامهء فرميينی برای تاشه های شبيه به شکل 91
(*) Formian scheme for configurations

   similar to Fig 91:  Ribbed dome.

   See Section 1.9 of Reference 1. (*)

S=36;      (*) Span (*)

H=6;       (*) Rise (*)

m=24;     (*) Frequency of

                                   elements on a ring (*)

n=9;   (*) Frequency of elements on a rib (*)

A=2*atan|(2*H/S);   (*) Sweep angle (*)

R=S/(2*sin|A);   (*) Radius (*)

E=rinit(m,n-1,1,1)|{[1,0,1; 1,1,1],

   [1,0,1; 1,0,2]}#rin(2,m,1)|{[1,0,0; 1,0,1],

   [1,0,n; 1,1,n]};

Dome=bs(R,360/m,A/n)|E;

use &,vm(2),vt(2),vh(R,R,3*R, 0,0,0, 0,0,1);

clear;   draw Dome;
برنامهء فرميينی برای تاشه های شبيه به شکل 92
(*) Formian scheme for configurations

   similar to Fig 92:  Trimmed ribbed dome.

   See Section 1.9 of Reference 1. (*)

S=36;     (*) Span (*)

H=6;      (*) Rise (*)

m=24;    (*) Frequency of elements

                          on a ring, must be even (*)

n=9;  (*) Frequency of elements on a rib (*)

A=2*atan|(2*H/S);       (*) Sweep angle (*)

R=S/(2*sin|A);   (*) Radius (*)   p=ric|(m/2);

E=rinit(m,n-1,1,1)|{[1,0,1; 1,1,1],

   [1,0,1; 1,0,2]}#rin(2,m,1)|{[1,0,0; 1,0,1],

   [1,0,n; 1,1,n]};

E=lux(rin(2,p,2)|[1,0,1])|E#

   rin(2,p,2)|[1,1,1; 1,3,1];

Dome=bs(R,360/m,A/n)|E;

use &,vm(2),vt(2),vh(R,R,3*R, 0,0,0, 0,0,1);

clear;   draw Dome;

برنامهء فرميينی برای تاشه های شبيه به شکل 93
(*) Formian scheme for configurations

   similar to Fig 93:  Schwedler dome.

   See Section 1.9 of Reference 1. (*)

S=36;     (*) Span (*)

H=6;       (*) Rise (*)

m=16;     (*) Frequency of

                                  elements on a ring (*)

n=8;  (*) Frequency of elements on a rib (*)

A=2*atan|(2*H/S);   (*) Sweep angle (*)

R=S/(2*sin|A);          (*) Radius (*)

E=rinit(m,n-1,1,1)|{[1,0,1; 1,1,1],

   [1,0,1; 1,0,2], [1,0,1; 1,1,2]}#rin(2,m,1)|

   {[1,0,0; 1,0,1], [1,0,n; 1,1,n]};

Dome=bs(R,360/m,A/n)|E;

use &,vm(2),vt(2),vh(R,R,3*R, 0,0,0, 0,0,1);

clear;   draw Dome;
برنامهء فرميينی برای تاشه های شبيه به شکل 94
(*) Formian scheme for configurations

   similar to Fig 94:  Trimmed Schwedler

   dome. See Section 1.9 of Reference 1. (*)

S=36;     (*) Span (*)

H=6;       (*) Rise (*)

m=16;     (*) Frequency of

            elements on a ring, must be even (*)

n=8;  (*) Frequency of elements on a rib (*)

A=2*atan|(2*H/S);   (*) Sweep angle (*)

R=S/(2*sin|A);   (*) Radius (*)   p=ric|(m/2);

E=rinit(m,n-1,1,1)|{[1,0,1; 1,1,1],

   [1,0,1; 1,0,2]}#rin(2,m,1)|{[1,0,0; 1,0,1],

   [1,0,n; 1,1,n]};

E=lux(rin(2,p+1,2)|[1,0,1])|E#rin(2,p,2)|

   [1,1,1; 1,3,1];

E=E#rin(2,p,2)|[1,1,1; 1,3,2]#

   rinit(m,n-2,1,1)|[1,0,2; 1,1,3];

Dome=bs(R,360/m,A/n)|E;

use &,vm(2),vt(2),vh(R,R,3*R, 0,0,0, 0,0,1);

clear;   draw Dome;

برنامهء فرميينی برای تاشه های شبيه به شکل 95
(*) Formian scheme for configurations

   similar to Fig 95:  Schwedler dome.

   See Section 1.9 of Reference 1. (*)

S=36;     (*) Span (*)

H=6;       (*) Rise (*)

m=16;     (*) Frequency of

            elements on a ring, must be even (*)

n=7;  (*) Frequency of elements on a rib (*)

A=2*atan|(2*H/S);   (*) Sweep angle (*)

R=S/(2*sin|A);          (*) Radius (*)

E=rinit(m,n,1,1)|{[1,0,1; 1,1,1],

   [1,0,1; 1,0,2]}#rin(2,m,1)|{[1,0,0; 1,0,1],

   [1,0,n+1; 1,1,n+1]}#

   rinit(ric|(m/2),n,2,1)|lam(2,1)|[1,0,1; 1,1,2];

Dome=bs(R,360/m,A/(n+1))|E;

use &,vm(2),vt(2),vh(R,R,3*R, 0,0,0, 0,0,1);

clear;   draw Dome;
برنامهء فرميينی برای تاشه های شبيه به شکل 96
(*) Formian scheme for configurations

   similar to Fig 96:  Schwedler dome.

   See Section 1.9 of Reference 1. (*)

S=36;     (*) Span (*)

H=6;       (*) Rise (*)

m=18;     (*) Frequency of

             elements on a ring, must be even (*)

n=9;        (*) Frequency of

               elements on a rib, must be odd (*)

A=2*atan|(2*H/S);   (*) Sweep angle (*)

R=S/(2*sin|A);          (*) Radius (*)

E=rinit(m,n-1,1,1)|{[1,0,1; 1,1,1],

  [1,0,1; 1,0,2]}#rin(2,m,1)|{[1,0,0; 1,0,1],

  [1,0,n; 1,1,n]}#rinit(ric|(m/2),tic|(n/2),2,2)|

  lamit(1,2)|[1,1,1; 1,2,2];

Dome=bs(R,360/m,A/n)|E;

use &,vm(2),vt(2),vh(R,R,3*R, 0,0,0, 0,0,1);

clear;   draw Dome;

برنامهء فرميينی برای تاشه های شبيه به شکل 97
(*) Formian scheme for configurations

   similar to Fig 97:  Schwedler dome.

   See Section 1.9 of Reference 1. (*)

S=36;     (*) Span (*)

H=6;       (*) Rise (*)

m=18;     (*) Frequency of

              elements on a ring, must be even (*)

n=8;        (*) Frequency of

               elements on a rib, must be even (*)

A=2*atan|(2*H/S);   (*) Sweep angle (*)

R=S/(2*sin|A);          (*) Radius (*)

E=rinit(m,n-1,1,1)|{[1,0,1; 1,1,1],

  [1,0,1; 1,0,2]}#rin(2,m,1)|{[1,0,0; 1,0,1],

  [1,0,n; 1,1,n]}#rinit(ric|(m/2),ric|(n/2),2,2)|

  {[1,1,1; 1,2,2], [1,1,2; 1,2,1]}#

  rinit(ric|(m/2),ric|(n/2)-1,2,2)|

  {[1,2,2; 1,3,3], [1,2,3; 1,3,2]};

Dome=bs(R,360/m,A/n)|E;

use &,vm(2),vt(2),vh(R,R,3*R, 0,0,0, 0,0,1);

clear;   draw Dome;
برنامهء فرميينی برای تاشه های شبيه به شکل 98
(*) Formian scheme for configurations

   similar to Fig 98:  Double layer ribbed

   canopy dome.

   See Section 1.9 of Reference 1. (*)

S=60;     (*) Span of top layer (*)

P=25;     (*) Gap diameter (*)

D=2;       (*) Depth (*)

A=40;     (*) Sweep angle (*)

m=24;     (*) Frequency of

                                  elements on a ring (*)

n=3;  (*) Frequency of elements on a rib (*)

R=S/(2*sin|A);          (*) Top radius (*)

Rb=R-D;                  (*) Bottom radius (*)

G=asin|(P/(2*R));     (*) Gap angle (*)

i=(A-G)/(2*n);          (*) increment (*)

T=rinit(m,n+1,2,2*i)|[R,0,G; R,2,G]#

   rinit(m,n,2,2*i)|[R,0,G; R,0,G+2*i];

B=rinit(m,n,2,2*i)|[Rb,1,G+i; Rb,3,G+i]#

   rinit(m,n-1,2,2*i)|[Rb,1,G+i; Rb,1,G+3*i];

W=rinit(m,n,2,2*i)|lamit(1,G+i)|

   [R,0,G; Rb,1,G+i];

Dome=bs(1,360/(2*m),1)|(T#B#W);

use &,vm(2),vt(2),vh(R,R,4*R, 0,0,0, 0,0,1);

clear;   draw Dome;
برنامهء فرميينی برای تاشه های شبيه به شکل 99
(*) Formian scheme for configurations

   similar to Fig 99:  Lamella dome.

   See Section 1.9 of Reference 1. (*)

S=36;     (*) Span (*)

H=5;       (*) Rise (*)

m=16;     (*) Frequency of

                     rhombuses in U2 direction (*)

n=5;        (*) Frequency of

                     rhombuses in U3 direction (*)

A=2*atan|(2*H/S);   (*) Sweep angle (*)

R=S/(2*sin|A);          (*) Radius (*)

E=rinit(m,n,2,2)|lamit(0,1)|[1,0,0; 1,1,1];

Dome=pex|bs(R,360/(2*m),A/n)|E;

use &,vm(2),vt(2),vh(R,R,4*R, 0,0,0, 0,0,1);

clear;   draw Dome;
برنامهء فرميينی برای تاشه های شبيه به شکل 100
(*) Formian scheme for configurations

   similar to Fig 100:  Lamella dome.

   See Section 1.9 of Reference 1. (*)

S=36;     (*) Span (*)

H=5;       (*) Rise (*)

m=16;     (*) Frequency of

                    rhombuses in U2 direction (*)

n=5;        (*) Frequency of

                    rhombuses in U3 direction (*)

A=2*atan|(2*H/S);   (*) Sweep angle (*)

R=S/(2*sin|A);          (*) Radius (*)

E=pex|rinit(m,n,2,2)|(lamit(0,1)|

   [1,0,0; 1,1,1]#[1,1,1; 1,-1,1]);

Dome=bs(R,360/(2*m),A/n)|E;

use &,vm(2),vt(2),vh(R,R,4*R, 0,0,0, 0,0,1);

clear;   draw Dome;
برنامهء فرميينی برای تاشه های شبيه به شکل 101
(*) Formian scheme for configurations

   similar to Fig 101:  Trimmed lamella dome.

   See Section 1.9 of Reference 1. (*)

S=36;     (*) Span (*)

H=4;       (*) Rise (*)

m=24;     (*) Frequency of rhombuses

              in U2 direction, m must be even (*)

n=6;        (*) Frequency of rhombuses

    in U3 direction, n must be larger than 3 (*)

p=ric|(m/2);

A=2*atan|(2*H/S);   (*) Sweep angle (*)

R=S/(2*sin|A);          (*) Radius (*)

E=rinit(m, n-3,2,2)|lamit(0,7)|[1,0,6; 1,1,7];

F=rinit(p,3,4,2)|lamit(0,1)|[1,0,0; 1,2,1];

G=E#F#rinit(m,2,2,2*(n-3))|[1,0,6; 1,2,6]#

   rin(2,p,4)|[1,2,6; 1,2,5];

Dome=pex|bs(R,360/(2*m),A/n)|G;

use &,vm(2),vt(2),vh(R,R,4*R, 0,0,0, 0,0,1);

clear;   draw Dome;
برنامهء فرميينی برای تاشه های شبيه به 

شکلهای 102 و 103
(*) Formian scheme for configurations

   similar to Fig 102 (f=1.2) and

   Fig 103 (f=1.4): Ovate dome.

   See Section 1.9 of Reference 1. (*)

S=36;     (*) Span (*)

H=8;      (*) Rise (*)

f=1.2;     (*) Factor for

                      scaling of half the dome (*)

A=2*atan|(2*H/S);   (*) Sweep angle (*)

R=S/(2*sin|A);          (*) Radius (*)

E1=rin(2,3,4)|{[1,0,0; 1,0,1], [1,0,1; 1,4,1],

   [1,0,1; 1,0,2], [1,0,1; 1,2,2],

   [1,2,2; 1,4,1]}#rin(2,6,2)|{[1,0,2; 1,2,2],

   [1,0,2; 1,0,3], [1,0,2; 1,2,3]}#

   rin(2,6,2)|{[1,0,3; 1,2,3], [1,0,3; 1,0,4],

   [1,0,3; 1,1,4], [1,1,4; 1,2,3]};

E2=rinit(12,3,1,1)|{[1,0,4; 1,1,4],

   [1,0,4; 1,0,5], [1,0,4; 1,1,5]};

E3=E1#E2#rin(2,12,1)|[1,0,7; 1,1,7];

E4=bs(R,360/24,A/7)|E3;

Dome=E4#bb(1,f)|refid (0,0)|E4;

use &,vm(2),vt(2),vh(R,0,3*R, 0,0,0, 0,0,1);

clear;   draw Dome;
برنامهء فرميينی برای تاشه های شبيه به شکل 104
(*) Formian scheme for configurations

   similar to Fig 104:  Diamatic dome.

   See Section 2.5 of Reference 2. (*)

R=36;      (*) Radius of circumsphere (*)

A=40;      (*) Sweep angle (*)

m=7;        (*) Frequency (*)

n=6;         (*) Number of sectors (*)

E=genit(1,m,1,1,0,1)|{[1,0,0; 1,0,1],

   [1,0,0; 1,1,1], [1,0,1; 1,1,1]};

F=bd(R,360/n,A/m)|E;

Dome=pex|rosad(0,0,n,360/n)|F;

use &,vm(2),vt(2),vh(R,R,3*R, 0,0,0, 0,0,1);

clear;   draw Dome;
برنامهء فرميينی برای تاشه های شبيه به شکل 105
(*) Formian scheme for configurations

   similar to Fig 105:  Diamatic dome.

   See Section 2.5 of Reference 2. (*)

R=36;      (*) Radius of circumsphere (*)

A=45;      (*) Sweep angle (*)

m=8;        (*) Frequency (*)

n=7;         (*) Number of sectors (*)

E=genit(1,m,3,3,0,1)|{[1,0,0; 1,1,2],

   [1,1,2; 1,0,3], [1,1,2; 1,3,3]};

F=bd(R,360/n,A/(3*m))|E;

Dome=rosad(0,0,n,360/n)|F;

use &,vm(2),vt(2),vh(R,R,3*R, 0,0,0, 0,0,1);

clear;   draw Dome;
برنامهء فرميينی برای تاشه های شبيه به شکل 106
(*) Formian scheme for configurations

   similar to Fig 106:  Diamatic dome.

   See Section 2.5 of Reference 2. (*)

R=36;      (*) Radius of circumsphere (*)

A=45;      (*) Sweep angle (*)

m=7;        (*) Frequency (*)

n=7;         (*) Number of sectors (*)

E=genit(1,m,3,3,0,1)|{[1,2,4; 1,1,2],

   [1,2,4; 1,1,5], [1,2,4; 1,4,5]}#

   rin(3,m+1,3)|[1,1,2; 1,-1,2];

F=bd(R,360/n,A/(3*m))|E;

Dome=rosad(0,0,n,360/n)|F;

use &,vm(2),vt(2),vh(R,R,3*R, 0,0,0, 0,0,1);

clear;   draw Dome;
برنامهء فرميينی برای تاشه های شبيه به شکل 107
(*) Formian scheme for configurations

   similar to Fig 107:  Diamatic dome.

   See Section 2.5 of Reference 2. (*)

R=36;      (*) Radius of circumsphere (*)

A=40;      (*) Sweep angle (*)

m=9;        (*) Frequency (*)

n=6;         (*) Number of sectors (*)

E=genit(1,m,1,1,0,1)|{[1,0,0; 1,0,1],

   [1,0,0; 1,1,1], [1,0,1; 1,1,1]};

Hole={[1,2,4], [1,2,7], [1,5,7]};

F=bd(R,360/n,A/m)|lux(Hole)|E;

Dome=pex|rosad(0,0,n,360/n)|F;

use &,vm(2),vt(2),vh(R,R,3*R, 0,0,0, 0,0,1);

clear;   draw Dome;
برنامهء فرميينی برای تاشه های شبيه به شکل 108
(*) Formian scheme for configurations

   similar to Fig 108:  Diamatic dome.

   See Section 2.5 of Reference 2. (*)

R=36;      (*) Radius of circumsphere (*)

A=45;      (*) Sweep angle (*)

m=8;        (*) Frequency (*)

n=7;         (*) Number of sectors (*)

E=genit(1,m,3,3,0,1)|{[1,0,0; 1,1,2],

   [1,1,2; 1,0,3], [1,1,2; 1,3,3]};

Hole=rin(2,5,3)|[1,6,24]#rin(2,2,3)|[1,9,21]#

   {[1,10,20], [1,7,23], [1,16,23]};

F=bd(R,360/n,A/(3*m))|lux(Hole)|E;

Dome=rosad(0,0,n,360/n)|F;

use &,vm(2),vt(2),vh(R,R,3*R, 0,0,0, 0,0,1);

clear;   draw Dome;
برنامهء فرميينی برای تاشه های شبيه به 

شکلهای 109 و 110
(*) Formian scheme for configurations

   similar to Fig 109 (f1=1.15, f2=0.85) and

   Fig 110 (f1=1.35, f2=1.05):

   Ovate diamatic dome.

   See Section 2.5 of Reference 2. (*)

R=36;  (*) Initial radius of circumsphere (*)

A=50;  (*) Sweep angle (*)

m=8;    (*) Frequency (*)

f1=1.15;   (*) Scale factor for 1st half (*)

f2=0.85;   (*) Scale factor for 2nd half (*)

n1=4;   (*) Number of sectors in 1st half (*)

n2=3;   (*) Number of sectors in 1st half (*)

E=genit(1,m,3,3,0,1)|{[1,0,0; 1,1,2],

   [1,1,2; 1,0,3], [1,1,2; 1,3,3]};

E1=bd(R,180/n1,A/(3*m))|E;

E2=bd(R,180/n2,A/(3*m))|E;

Dome=dil(2,f1)|rosad(0,0,n1,180/n1)|E1#

   dil(2,f2)|ref(2,0)|rosad(0,0,n2,180/n2)|E2;

use &,vm(2),vt(2),vh(R,0,3*R, 0,0,0, 0,0,1);

clear;   draw Dome;
برنامهء فرميينی برای تاشه های شبيه به شکل 111
(*) Formian scheme for configurations

   similar to Fig 111:  Double layer diamatic

   dome. See Section 2.5 of Reference 2. (*)

R=36;     (*) Radius of top circumsphere (*)

Rb=34.5;  (*) Radius of

                             bottom circumsphere (*)

A=45;      (*) Sweep angle (*)

m=3;        (*) Frequency  of top layer (*)

n=6;         (*) Number of sectors (*)

T=genit(1,m,3,3,0,1)|{[R,0,0; R,0,3],

   [R,0,0; R,3,3], [R,0,3; R,3,3]};

B=genit(1,m-1,3,3,0,1)|{[Rb,2,4; Rb,1,2],

   [Rb,2,4; Rb,1,5], [Rb,2,4; Rb,4,5]}#

   rin(3,m,3)|[Rb,1,2; Rb,-1,2];

W=genit(1,m,3,3,0,1)|{[Rb,1,2; R,0,0],

   [Rb,1,2; R,0,3], [Rb,1,2; R,3,3]}#

   genit(1,m-1,3,3,0,1)|{[Rb,2,4; R,3,6],

   [Rb,2,4; R,0,3], [Rb,2,4; R,3,3]};

F=bd(1,360/n,A/(3*m))|(T#B#W);

Dome=pex|rosad(0,0,n,360/n)|F;

use &,vm(2),vt(2),vh(R,R,3*R, 0,0,0, 0,0,1);

clear;   draw Dome;
برنامهء فرميينی برای تاشه های شبيه به شکل 112
(*) Formian scheme for configurations

   similar to Fig 112:  Scallop dome.

   See Reference 12. (*)

R=36;      (*) Radius of circumsphere (*)

A=45;      (*) Sweep angle (*)

m=8;        (*) Frequency (*)

n=6;         (*) Number of sectors (*)

C=1;        (*) Code (*)

D=9;        (*) Amplitude (*)

P=0;         (*) Prominence (*)

nn=360/n;

E=genit(1,m,1,1,0,1)|{[1,0,0; 1,0,1],

   [1,0,0; 1,1,1], [1,0,1; 1,1,1]};

F=sbd(C,nn,[R,m,D,P],1,nn,A/m)|E;

Dome=pex|rosad(0,0,n,nn)|F;

use &,vm(2),vt(2),vh(R,R,2*R, 0,0,0, 0,0,1);

clear;   draw Dome;
برنامهء فرميينی برای تاشه های شبيه به شکل 113
(*) Formian scheme for configurations

   similar to Fig 113:  Scallop dome.

   See Reference 12. (*)

R=36;      (*) Radius of circumsphere (*)

A=70;      (*) Sweep angle (*)

m=7;        (*) Frequency (*)

n=8;         (*) Number of sectors (*)

C=1;        (*) Code (*)

D=15;      (*) Amplitude (*)

P=0;         (*) Prominence (*)

nn=360/n;

E=genit(1,m,1,1,0,1)|{[1,0,0; 1,0,1],

   [1,0,0; 1,1,1], [1,0,1; 1,1,1]};

F=sbd(C,nn,[R,m,D,P],1,nn,A/m)|E;

Dome=pex|rosad(0,0,n,nn)|F;

use &,vm(2),vt(2),vh(R,R,3*R, 0,0,0, 0,0,1);

clear;   draw Dome;
برنامهء فرميينی برای تاشه های شبيه به شکل 114
(*) Formian scheme for configurations

   similar to Fig 114:  Scallop dome.

   See Reference 12. (*)

R=36;      (*) Radius of circumsphere (*)

A=70;      (*) Sweep angle (*)

m=8;        (*) Frequency (*)

n=5;         (*) Number of sectors (*)

C=1;        (*) Code (*)

D=8;      (*) Amplitude (*)

P=15;       (*) Prominence (*)

nn=360/n;

E=genit(1,m,1,1,0,1)|{[1,0,0; 1,0,1],

   [1,0,0; 1,1,1], [1,0,1; 1,1,1]};

F=sbd(C,nn,[R,m,D,P],1,nn,A/m)|E;

Dome=pex|rosad(0,0,n,nn)|F;

use &,vm(2),vt(2),vh(R,R,3*R, 0,0,0, 0,0,1);

clear;   draw Dome;
برنامهء فرميينی برای تاشه های شبيه به شکل 115
(*) Formian scheme for configurations

   similar to Fig 115:  Scallop dome.

   See Reference 12. (*)

R=36;      (*) Radius of circumsphere (*)

A=32;      (*) Sweep angle (*)

m=7;        (*) Frequency (*)

n=12;       (*) Number of sectors (*)

C=1;        (*) Code (*)

D=4;        (*) Amplitude (*)

P=0;        (*) Prominence (*)

nn=360/n;

E=genit(1,m,1,1,0,1)|{[1,0,0; 1,0,1],

   [1,0,0; 1,1,1], [1,0,1; 1,1,1]};

F=sbd(C,nn,[R,m,D,P],1,nn,A/m)|E;

Dome=pex|rosad(0,0,n,nn)|F;

use &,vm(2),vt(2),vh(R,R,2*R, 0,0,0, 0,0,1);

clear;   draw Dome;

برنامهء فرميينی برای تاشه های شبيه به شکل 116
(*) Formian scheme for configurations

   similar to Fig 116:  Scallop dome.

   See Reference 12. (*)

R=36;      (*) Radius of circumsphere (*)

A=45;      (*) Sweep angle (*)

m=8;        (*) Frequency (*)

n=6;         (*) Number of sectors (*)

C=2;        (*) Code (*)

D=10;      (*) Amplitude (*)

P=0;         (*) Prominence (*)

nn=360/n;

E=genit(1,m,1,1,0,1)|{[1,0,0; 1,0,1],

   [1,0,0; 1,1,1], [1,0,1; 1,1,1]};

F=sbd(C,nn,[R,m,D,P],1,nn,A/m)|E;

Dome=pex|rosad(0,0,n,nn)|F;

use &,vm(2),vt(2),vh(R,R,2*R, 0,0,0, 0,0,1);

clear;   draw Dome;

برنامهء فرميينی برای تاشه های شبيه به شکل 117
(*) Formian scheme for configurations

   similar to Fig 117:  Scallop dome.

   See Reference 12. (*)

R=36;      (*) Radius of circumsphere (*)

A=60;      (*) Sweep angle (*)

m=7;        (*) Frequency (*)

n=8;         (*) Number of sectors (*)

C=2;        (*) Code (*)

D=6;        (*) Amplitude (*)

P=0;        (*) Prominence (*)

nn=360/n;

E=genit(1,m,1,1,0,1)|{[1,0,0; 1,0,1],

   [1,0,0; 1,1,1], [1,0,1; 1,1,1]};

F=sbd(C,nn,[R,m,D,P],1,nn,A/m)|E;

Dome=pex|rosad(0,0,n,nn)|F;

use &,vm(2),vt(2),vh(R,R,2*R, 0,0,0, 0,0,1);

clear;   draw Dome;
برنامهء فرميينی برای تاشه های شبيه به شکل 118
(*) Formian scheme for configurations

   similar to Fig 118:  Scallop dome.

   See Reference 12. (*)

R=36;      (*) Radius of circumsphere (*)

A=45;      (*) Sweep angle (*)

m=11;      (*) Frequency (*)

n=3;         (*) Number of sectors (*)

C=2;        (*) Code (*)

D=5;        (*) Amplitude (*)

P=20;        (*) Prominence (*)

nn=360/n;

E=genit(1,m,1,1,0,1)|{[1,0,0; 1,0,1],

   [1,0,0; 1,1,1], [1,0,1; 1,1,1]};

F=sbd(C,nn,[R,m,D,P],1,nn,A/m)|E;

Dome=pex|rosad(0,0,n,nn)|F;

use &,vm(2),vt(2),vh(R,R,3*R, 0,0,0, 0,0,1);

clear;   draw Dome;
برنامهء فرميينی برای تاشه های شبيه به شکل 119
(*) Formian scheme for configurations

   similar to Fig 119:  Scallop dome.

   See Reference 12. (*)

R=36;      (*) Radius of circumsphere (*)

A=70;      (*) Sweep angle (*)

m=10;      (*) Frequency (*)

n=2;         (*) Number of sectors (*)

C=2;        (*) Code (*)

D=8;        (*) Amplitude (*)

P=30;       (*) Prominence (*)

nn=360/n;

E=genit(1,m,1,1,0,1)|{[1,0,0; 1,0,1],

   [1,0,0; 1,1,1], [1,0,1; 1,1,1]};

F=sbd(C,nn,[R,m,D,P],1,nn,A/m)|E;

Dome=pex|rosad(0,0,n,nn)|F;

use &,vm(2),vt(2),vh(R,R,4*R, 0,0,0, 0,0,1);

clear;   draw Dome;
برنامهء فرميينی برای تاشه های شبيه به شکل 120
(*) Formian scheme for configurations

   similar to Fig 120:  Onion dome.

   See Section 3.4 of Reference 3. (*)

R=10;       (*) First radius (*)

Q=50;       (*) Second radius (*)

B=40;       (*) Base angle (*)

m=24;   (*) Number of rhombi along U2 (*)

n=8;      (*) Number of rhombi along U3 (*)

V=1;     (*) Vertical scale factor (*)

A=acos|(R/Q)+B;      (*) Sweep angle (*)

P=asin|(R/Q);             (*) Position angle (*)

E=rinic(2,4,m,n,2,-2)|lamic(2,4,1,-1)|

   [1,1,1,0; 1,0,1,-1]#rin(2,m,2)|

   [1,1,1,-2*n; 1,3,1,-2*n];

F=tran(4,-2*n*P/A)|E;

G=pex|dep(4)|ba(R,360/(2*m),Q,A/(2*n))|F;

Dome=dil(3,V)|G;

use &,vm(2),vt(2),

   vh(0,100000*Q,Q, 0,0,0, 0,0,1);

clear;   draw Dome;
برنامهء فرميينی برای تاشه های شبيه به شکل 121
(*) Formian scheme for configurations

   similar to Fig 121:  Onion dome.

   See Section 3.4 of Reference 3. (*)

R=10;      (*) First radius (*)

Q=50;      (*) Second radius (*)

B=20;       (*) Base angle (*)

m=18;   (*) Number of rhombi along U2 (*)

n=8;      (*) Number of rhombi along U3 (*)

V=2.5;  (*) Vertical scale factor (*)

A=acos|(R/Q)+B;      (*) Sweep angle (*)

P=asin|(R/Q);             (*) Position angle (*)

E=rinic(2,4,m,n,2,-2)|lamic(2,4,1,-1)|

   [1,1,1,0; 1,0,1,-1]#rin(2,m,2)|

   [1,1,1,-2*n; 1,3,1,-2*n];

F=tran(4,-2*n*P/A)|E;

G=pex|dep(4)|ba(R,360/(2*m),Q,A/(2*n))|F;

Dome=dil(3,V)|G;

use &,vm(2),vt(2),

   vh(0,100000*Q,Q, 0,0,0, 0,0,1);

clear;   draw Dome;
برنامهء فرميينی برای تاشه های شبيه به شکل 122
(*) Formian scheme for configurations

   similar to Fig 122:  Onion dome.

   See Section 3.4 of Reference 3. (*)

R=10;      (*) First radius (*)

Q=65;      (*) Second radius (*)

B=60;      (*) Base angle (*)

m=20;   (*) Number of rhombi along U2 (*)

n=8;      (*) Number of rhombi along U3 (*)

V=2.5;  (*) Vertical scale factor (*)

A=acos|(R/Q)+B;      (*) Sweep angle (*)

P=asin|(R/Q);             (*) Position angle (*)

E=rinic(2,4,m,n,2,-2)|lamic(2,4,1,-1)|

   [1,1,1,0; 1,0,1,-1]#rin(2,m,2)|

   [1,1,1,-2*n; 1,3,1,-2*n];

F=tran(4,-2*n*P/A)|E;

G=pex|dep(4)|ba(R,360/(2*m),Q,A/(2*n))|F;

Dome=dil(3,V)|G;

use &,vm(2),vt(2),

   vh(0,100000*Q,Q, 0,0,0, 0,0,1);

clear;   draw Dome;
برنامهء فرميينی برای تاشه های شبيه به شکل 123
(*) Formian scheme for configurations

   similar to Fig 123:  Onion dome.

   See Section 3.4 of Reference 3. (*)

R=10;      (*) First radius (*)

Q=50;      (*) Second radius (*)

B=40;      (*) Base angle (*)

m=24;   (*) Number of rhombi along U2 (*)

n=8;      (*) Number of rhombi along U3 (*)

V=0.5;  (*) Vertical scale factor (*)

A=acos|(R/Q)+B;      (*) Sweep angle (*)

P=asin|(R/Q);             (*) Position angle (*)

E=rinic(2,4,m,n,2,-2)|lamic(2,4,1,-1)|

   [1,1,1,0; 1,0,1,-1]#rin(2,m,2)|

   [1,1,1,-2*n; 1,3,1,-2*n];

F=tran(4,-2*n*P/A)|E;

G=pex|dep(4)|ba(R,360/(2*m),Q,A/(2*n))|F;

Dome=dil(3,V)|G;

use &,vm(2),vt(2),

   vh(0,100000*Q,Q, 0,0,0, 0,0,1);

clear;   draw Dome;
برنامهء فرميينی برای تاشه های شبيه به شکل 124
(*) Formian scheme for configurations

   similar to Fig 124: Upcurved mallow
   dome. See Section 3.2 of Reference 3. (*)

L=40;        (*) Span of dome (*)

C=20;       (*) Central height (*)

P=0.04;     (*) Pitch (*)

T=-1;        (*) Curving style coefficient (*)

m=7;         (*) Frequency (*)

n=7;          (*) Number of sectors (*)

B=360/(2*n);

D=sqrt|(cos|B^2*L*L/4+C*C);

        (*)  D must be lager than C  (*)

A=acos|(C/D);   S=D/m;

W=D*sin|A*tan|B;   K=W/m;

E1=genid(1,m,2,S,-1,1)|{[0,0,0; -1,S,0],

   [0,0,0; 1,S,0], [-1,S,0; 1,S,0]};

E2=verat(0,0,A-90)|E1;

E3=dil(1,K)|E2;

E4=verad(0,0,T*45)|E3;

F=bhp(1,1,1,P)|E4;

Dome=pex|rosad(0,0,n,360/n)|F;

use &,vm(2),vt(2),

   vh(L,L,1.3*L, 0,0,0, 0,0,1);

clear;   draw Dome;
برنامهء فرميينی برای تاشه های شبيه به شکل 125
(*) Formian scheme for configurations

   similar to Fig 125:  Upcurved mallow
   dome. See Section 3.2 of Reference 3. (*)

L=40;        (*) Span of dome (*)

C=6;         (*) Central height (*)

P=0.04;     (*) Pitch (*)

T=-1;        (*) Curving style coefficient (*)

m=9;         (*) Frequency (*)

n=3;          (*) Number of sectors (*)

B=360/(2*n);

D=sqrt|(cos|B^2*L*L/4+C*C);

        (*)  D must be lager than C  (*)

A=acos|(C/D);   S=D/m;

W=D*sin|A*tan|B;   K=W/m;

E1=genid(1,m,2,S,-1,1)|{[0,0,0; -1,S,0],

   [0,0,0; 1,S,0], [-1,S,0; 1,S,0]};

E2=verat(0,0,A-90)|E1;

E3=dil(1,K)|E2;

E4=verad(0,0,T*45)|E3;

F=bhp(1,1,1,P)|E4;

Dome=pex|rosad(0,0,n,360/n)|F;

use &,vm(2),vt(2),vh(L,L,2*L, 0,0,0, 0,0,1);

clear;   draw Dome;
برنامهء فرميينی برای تاشه های شبيه به شکل 126
(*) Formian scheme for configurations

   similar to Fig 126:  Upcurved mallow
   dome. See Section 3.2 of Reference 3. (*)

L=36;        (*) Span of dome (*)

C=20;       (*) Central height (*)

P=0.05;     (*) Pitch (*)

T=-1;        (*) Curving style coefficient (*)

m=7;         (*) Frequency (*)

n=10;        (*) Number of sectors (*)

B=360/(2*n);

D=sqrt|(cos|B^2*L*L/4+C*C);

        (*)  D must be lager than C  (*)

A=acos|(C/D);   S=D/m;

W=D*sin|A*tan|B;   K=W/m;

E1=genid(1,m,2,S,-1,1)|{[0,0,0; -1,S,0],

   [0,0,0; 1,S,0], [-1,S,0; 1,S,0]};

E2=verat(0,0,A-90)|E1;

E3=dil(1,K)|E2;

E4=verad(0,0,T*45)|E3;

F=bhp(1,1,1,P)|E4;

Dome=pex|rosad(0,0,n,360/n)|F;

use &,vm(2),vt(2),

   vh(L,L,1.7*L, 0,0,0, 0,0,1);

clear;   draw Dome;
برنامهء فرميينی برای تاشه های شبيه به شکل 127
(*) Formian scheme for configurations

   similar to Fig 127:  Downcurved mallow

   dome. See Section 3.2 of Reference 3. (*)

L=36;        (*) Span of dome (*)

C=1.5;       (*) Central height (*)

P=0.03;     (*) Pitch (*)

T=1;          (*) Curving style coefficient (*)

m=7;          (*) Frequency (*)

n=8;           (*) Number of sectors (*)

B=360/(2*n);

D=sqrt|(cos|B^2*L*L/4+C*C);

        (*)  D must be lager than C  (*)

A=acos|(C/D);   S=D/m;

W=D*sin|A*tan|B;   K=W/m;

E1=genid(1,m,2,S,-1,1)|{[0,0,0; -1,S,0],

   [0,0,0; 1,S,0], [-1,S,0; 1,S,0]};

E2=verat(0,0,A-90)|E1;

E3=dil(1,K)|E2;

E4=verad(0,0,T*45)|E3;

F=bhp(1,1,1,P)|E4;

Dome=pex|rosad(0,0,n,360/n)|F;

use &,vm(2),vt(2),vh(L,L,2*L, 0,0,0, 0,0,1);

clear;   draw Dome;
برنامهء فرميينی برای تاشه های شبيه 
به شکلهای 128 و 129
(*) Formian scheme for configurations

   similar to Fig 128 (n=4) and Fig 129 (n=3):

   Downcurved mallow dome.

   See Section 3.2 of Reference 3. (*)

L=36;        (*) Span of dome (*)

C=4.5;       (*) Central height (*)

P=0.03;     (*) Pitch (*)

T=1;          (*) Curving style coefficient (*)

m=8;          (*) Frequency (*)

n=4;           (*) Number of sectors (*)

B=360/(2*n);

D=sqrt|(cos|B^2*L*L/4+C*C);

       (*)  D must be lager than C  (*)

A=acos|(C/D);   S=D/m;

W=D*sin|A*tan|B;   K=W/m;

E1=genid(1,m,2,S,-1,1)|{[0,0,0; -1,S,0],

   [0,0,0; 1,S,0], [-1,S,0; 1,S,0]};

E2=verat(0,0,A-90)|E1;

E3=dil(1,K)|E2;

E4=verad(0,0,T*45)|E3;

F=bhp(1,1,1,P)|E4;

Dome=pex|rosad(0,0,n,360/n)|F;

use &,vm(2),vt(2),

   vh(L,3*L,6*L, 0,0,0, 0,0,1);

clear;   draw Dome;
برنامهء فرميينی برای تاشه های شبيه به شکل 130

(*) Formian scheme for configurations

   similar to Fig 130:  Pellevated dome

   (spherical) with binate pattern.
   See References 4, 5 and 6. (*)

L1=25;  (*) Length in U1 direction (*)

L2=25;  (*) Length in U2 direction (*)

m=12;   (*) Frequency in U1 direction (*)

n=12;    (*) Frequency in U2 direction (*)

S=sqrt|(L1^2+L2^2);

                        (*) Span of pellevant (*)

R=0.3*S;         (*) Rise of pellevant (*)

E=rinid(m,n+1,1,1)|[0,0,0; 1,0,0]#

   rinid(m+1,n,1,1)|[0,0,0; 0,1,0];

Grid=bb(L1/m,L2/n)|E;

Dome=capel(1,L1/2,L2/2,S,R)|Grid;

use &,vm(2),vt(2),vh(L1/2,-L2,3*L2,

   L1/2,L2/2,0, L1/2,L2/2,1);

clear;   draw Dome;

برنامهء فرميينی برای تاشه های شبيه به شکل 131

(*) Formian scheme for configurations

   similar to Fig 131:  Pellevated dome

   (parabolic) with obnate pattern.

   See References 4, 5 and 6. (*)
L1=25;  (*) Length in U1 direction (*)

L2=25;  (*) Length in U2 direction (*)

m=7;     (*) Frequency of crosses

                                     in U1 direction (*)

n=7;      (*) Frequency of crosses

                                     in U2 direction (*)

S=sqrt|(L1^2+L2^2);

                        (*) Span of pellevant (*)

R=0.35*S;       (*) Rise of pellevant (*)

E=rinid(m,n,2,2)|lamid(1,1)|[0,0,0; 1,1,0]#

   rinid(m,2,2,2*n)|[0,0,0; 2,0,0]#

   rinid(2,n,2*m,2)|[0,0,0; 0,2,0];

Grid=bb(L1/(2*m),L2/(2*n))|E;

Dome=capel(3,L1/2,L2/2,S,R)|Grid;

use &,vm(2),vt(2),vh(L1/2,-L2,3*L2,

   L1/2,L2/2,0, L1/2,L2/2,1);

clear;   draw Dome;

برنامهء فرميينی برای تاشه های شبيه به شکل 132

(*) Formian scheme for configurations

   similar to Fig 132:  Compound pellevated

   dome (parabolic) with obnate pattern.

   See References 4, 5 and 6. (*)

L1=25;  (*) Length in U1 direction (*)

L2=25;  (*) Length in U2 direction (*)

m=6;     (*) Frequency of crosses

                                     in U1 direction (*)

n=6;      (*) Frequency of crosses

                                     in U2 direction (*)

S=sqrt|(L1^2+L2^2);

                        (*) Span of pellevant (*)

R=0.25*S;       (*) Rise of pellevant (*)

E=rinid(m,n,2,2)|lamid(1,1)|[0,0,0; 1,1,0]#

   rinid(m,2,2,2*n)|[0,0,0; 2,0,0]#

   rinid(2,n,2*m,2)|[0,0,0; 0,2,0];

Grid=bb(L1/(2*m),L2/(2*n))|E;

D=capel(3,L1/2,L2/2,S,R)|Grid;

Dome=pex|rinid(2,2,L1,L2)|D;

use &,vm(2),vt(2),vh(L1/2,-L2,3*L2,

   L1/2,L2/2,0, L1/2,L2/2,1);

clear;   draw Dome;

برنامهء فرميينی برای تاشه های شبيه به شکل 133
(*) Formian scheme for configurations

   similar to Fig 133:  Pellevated dome

   (conical) with ternate pattern.

   See References 4, 5 and 6. (*)

L=24;  (*) Length of each side of dome (*)

m=12;  (*) Frequency (*)

Grid=bb(L/(2*m),L*(sqrt|3)/(2*m))|

   genid(m,m,2,1,1,-1)|{[0,0,0; 2,0,0],

   [1,1,0; 0,0,0], [1,1,0; 2,0,0]};

Dome=capel(4,L/2,S/4,S,R)|Grid;
use &,vm(2),vt(2),vh(L/2,-L,2*L,

   L/2,L/2,0, L/2,L/2,1);

clear;   draw Dome;

برنامهء فرميينی برای تاشه های شبيه به شکل 134
(*) Formian scheme for configurations

   similar to Fig 134:  Pellevated sectorate

   dome (parabolic).
   See References 4, 5 and 6. (*)

S=40;        (*) Span of dome (*)

H=10;        (*) Rise (*)

m=7;          (*) Frequency (*)

n=6;           (*) Number of sides (*)

nn=180/n;   L=S*sin|nn;

E=genid(1,m,2,1,-1,1)|{[0,0,0; -1,1,0],

   [0,0,0; 1,1,0], [-1,1,0; 1,1,0]};

F=bt(L/(2*m),L/(2*m*tan|nn),1)|E;

Base=pex|rosad(0,0,n,360/n)|F;

Dome=capel(3,0,0,S,H)|Base;

use &,vm(2),vt(2),vh(0,S,2*S, 0,0,0, 0,0,1);

clear;   draw Dome;
برنامهء فرميينی برای تاشه های شبيه 

به شکلهای 135 و 136
(*) Formian scheme for configurations

   similar to Fig 135 (n=5) and Fig 136 (n=6):
   Pellevated sectorate dome (spherical) with

   ternate/hexagonal pattern.

   See References 4, 5 and 6. (*)

L=20;  (*) Length of each side of the grid (*)

m=11;  (*) Frequency of the original ternate 

    grid, m must be a multiple of 3 minus 1(*)

n=5;     (*) Number of sides (*)

S=L/(sin|(360/(2*n)));
                           (*) Span of pellevant (*)

R=0.25*S;          (*) Rise of pellevant (*)

q=ric|((m+1)/3);

E=[-1,1,0; 1,1,0]#lam(1,0)|{[-1,1,0; 0,2,0],

   [-2,2,0; 0,2,0]};

F=genid(1,q,6,3,-3,1)|E#

   genid(1,q-1,6,3,-3,1)|(lamid(0,3)|lam(1,-1)|

   [-2,2,0; -1,3,0]#ref(2,3)|E);

Grid=rosad(0,0,n,360/n)|bb(L/(2*m),

   L/(2*m*tan|(180/n)))|F;

Dome=capel(1,0,0,S,R)|Grid;
use &,vm(2),vt(2),vh(0,-2*L,4*L,

   0,0,0, 0,0,1);

clear;   draw Dome;
برنامهء فرميينی برای تاشه های شبيه 

به شکل 137
(*) Formian scheme for configurations

   similar to Fig 137:  Pellevated sectorate

   dome (parabolic) with honeycomb pattern.

   See References 4, 5 and 6. (*)
L=20;  (*) Length of each side of the grid (*)

m=9;    (*) Frequency of the original ternate

                grid, m must be a multiple of 3 (*)

n=8;     (*) Number of sides (*)

S=L/(sin|(360/(2*n)));

                           (*) Span of pellevant (*)

R=0.35*S;          (*) Rise of pellevant (*)

E=genid(1,m,2,1,-1,1)|{[0,0,0; -1,1,0],

   [0,0,0; 1,1,0], [-1,1,0; 1,1,0]};

RN=genid(1,((m/3)+1),6,3,-3,1)|[0,0,0]#

   genid(1,m/3,6,3,-3,1)|[0,2,0]#

   genid(1,((m/3)-1),6,3,-3,1)|[0,4,0];

Grid=pex|rosad(0,0,n,360/n)|bb(L/(2*m),

   L/(2*m*tan|(180/n)))|lux(RN)|E;

Dome=capel(3,0,0,S,R)|Grid;

use &,vm(2),vt(2),vh(0,-2*L,5*L,

   0,0,0, 0,0,1);

clear;   draw Dome;

برنامهء فرميينی برای تاشه های شبيه به شکل 138
(*) Formian scheme for configurations

   similar to Fig 138:  Pellevated sectorate

   dome (tri-spherical). See References
   4, 5 and 6. (*)

S=40;        (*) Span of dome (*)

H=4;         (*) Rise (*)

m=7;         (*) Frequency (*)

n=6;          (*) Number of sides (*)

nn=180/n;   L=S*sin|nn;

E=genid(1,m,2,1,-1,1)|{[0,0,0; -1,1,0],

   [0,0,0; 1,1,0], [-1,1,0; 1,1,0]};

F=bt(L/(2*m),L/(2*m*tan|nn),1)|E;

Base=pex|rosad(0,0,n,360/n)|F;

Dome=capel(1,S*sin|nn/4,S*cos|nn/4,S,H)|

   capel(1,S*sin|nn/4,-S*cos|nn/4,S,H)|

   capel(1,-S*sin|nn/4,-S*cos|nn/4,S,H)|Base;

use &,vm(2),vt(2),vh(0,S,S, 0,0,0, 0,0,1);

clear;   draw Dome;
برنامهء فرميينی برای تاشه های شبيه به شکل 139
(*) Formian scheme for configurations

   similar to Fig 139:  Pellevated dome
   (parabolic + conical). See References
   4, 5 and 6. (*)

S=40;     (*) Span of dome (*)

H1=6;     (*) Rise of the paraboloid (*)

H2=4;     (*) Rise of the cone (*)

m=6;       (*) Frequency (*)

n=8;        (*) Number of sides (*)

nn=180/n;   L=S*sin|nn;

E=genid(1,m,2,1,-1,1)|{[0,0,0; -1,1,0],

   [0,0,0; 1,1,0], [-1,1,0; 1,1,0]};

F=bt(L/(2*m),L/(2*m*tan|nn),1)|E;

Base=pex|rosad(0,0,n,360/n)|F;

Dome=capel(3,0,0,S,H1)|

   capel(4,0,0,S,H2)|Base;

use &,vm(2),vt(2),

   vh(S,0,1.4*S, 0,0,0, 0,0,1);

clear;   draw Dome;
برنامهء فرميينی برای تاشه های شبيه به شکل 140
(*) Formian scheme for configurations

   similar to Fig 140:  Pellevated double layer
   dome (spherical). See References
   4, 5 and 6. (*)

L1=20;     (*) Length in U1 direction (*)

L2=20;     (*) Length in U2 direction (*)

D=1.5;      (*) Depth in U3 direction (*)

H=6;         (*) Rise of dome (*)

m=6;   (*) Frequency in U1 direction (*)

n=6;    (*) Frequency in U2 direction (*)

T=rinid(m,n+1,2,2)|[0,0,1; 2,0,1]#

   rinid(m+1,n,2,2)|[0,0,1; 0,2,1];

B=rinid(m-1,n,2,2)|[1,1,0; 3,1,0]#

   rinid(m,n-1,2,2)|[1,1,0; 1,3,0];

W=rinid(m,n,2,2)|lamid(1,1)|[0,0,1; 1,1,0];

G=bt(L1/(2*m),L2/(2*n),D)|(T#B#W);

Dome=capel(1,L1/2,L2/2,

   sqrt|(L1^2+L2^2),H)|G;

use &,vm(2),vt(2),vh(L1/2,-L2/2,2*L2,

   L1/2,L2/2,0, L1/2,L2/2,1);

clear;   draw Dome;
برنامهء فرميينی برای تاشه های شبيه به شکل 141
(*) Formian scheme for configurations

   similar to Fig 141:  Pellevated double layer
   dome (parabolic): See References
   4, 5 and 6. (*)

L=20;     (*) Length of each side in plan (*)

H=8;      (*) Rise of dome (*)

D=2.15; (*) Depth of grid (*)

m=3;      (*) Frequency of top layer (*)

n=8;       (*) Number of sides (*)

S=L/tan|(180/n);  (*) Span (*)

ET=genid(1,m,2,3,-1,1)|{[0,0,0; -1,3,0],


[0,0,0; 1,3,0], [-1,3,0; 1,3,0]};

EB=genid(1,m-1,2,3,-1,1)|{[0,2,-1; -1,5,-1],


[0,2,-1; 1,5,-1], [-1,5,-1; 1,5,-1]};

EW=genid(1,m,2,3,-1,1)|{[0,0,0; 0,2,-1],


[-1,3,0; 0,2,-1], [1,3,0; 0,2,-1]};

E=bt(L/(2*m),S/(6*m),D)|(ET#EB#EW);

Qi=[0,2,-1];   Ri=bt(L/(2*m),S/(6*m),D)|Qi;

Rj=verad(0,0,360/n)|Ri;   K1=tig|(Ri#Rj);

K2=rinad(0,0,-L/(2*m),S/(2*m),m)|K1;

Grid=pex|rosad(0,0,n,360/n)|(E#K2);

Dome=capel(3,0,0,S,H)|Grid;

use &,vm(2),vt(2),vh(0,S,2*S, 0,0,0, 0,0,1);

clear;   draw Dome;
برنامهء فرميينی برای تاشه های شبيه به شکل 142
(*) Formian scheme for configurations

   similar to Fig 142:  Geodesic dome,

   based on top 5 faces of icosahedron.

   See References 7 and 8. (*)

S=60;   (*) Span (*)

m=7;    (*) Frequency of subdivision (*)

R=sqrt|(10+2*sqrt|5)*sin|36*S/4;

E=pol(1,5,'[1-5]',R,[0,0; 2*m,0])|

   dil(2,sqrt|3)|genid(m,m,2,1,1,-1)|

   {[0,0; 2,0], [0,0; 1,1], [2,0; 1,1]};

Dome=trac(1,[0,0,0,R],4,-15)|E;

use &,vm(2),vt(2),vh(S,S,3*S, 0,0,0, 0,0,1);

clear;   draw Dome;

برنامهء فرميينی برای تاشه های شبيه به شکل 143
(*) Formian scheme for configurations

   similar to Fig 143:  Geodesic dome,

   based on top 5 faces of icosahedron.

   See References 7 and 8. (*)

S=60;   (*) Span (*)

m=12;  (*) Frequency,

                       m must be divisible by 3 (*)

mm=m/3-1;

R=sqrt|(10+2*sqrt|5)*sin|36*S/4;

E=genid(m,m,2,1,1,-1)|{[0,0; 2,0],

   [0,0; 1,1], [2,0; 1,1]};
E=lux(genid(mm,mm,6,3,3,-1)|[6,2])|E;

G=pol(1,5,'[1-5]',R,[0,0; 2*m,0])|
   dil(2,sqrt|3)|E;

Dome=trac(1,[0,0,0,R],4,-15)|G;

use &,vm(2),vt(2),vh(S,S,3*S, 0,0,0, 0,0,1);

clear;   draw Dome;
برنامهء فرميينی برای تاشه های شبيه به شکل 144
(*) Formian scheme for configurations

   similar to Fig 144:  Geodesic dome,

   based on top 5 faces of icosahedron.

   See References 7 and 8. (*)

S=60;   (*) Span (*)

m=14;  (*) Frequency, m must be even (*)

p=tic|(m/2)+1;

R=sqrt|(10+2*sqrt|5)*sin|36*S/4;

E=genid(m,m,2,1,1,-1)|{[0,0; 2,0],

   [0,0; 1,1], [2,0; 1,1]};

C=genid(p,p,4,2,2,-1)|[0,0];

G=pol(1,5,'[1-5]',R,[0,0; 2*m,0])|
   dil(2,sqrt|3)|lux(C)|E;

Dome=trac(1,[0,0,0,R],4,-15)|G;

use &,vm(2),vt(2),vh(S,S,3*S, 0,0,0, 0,0,1);

clear;   draw Dome;
برنامهء فرميينی برای تاشه های شبيه به شکل 145
(*) Formian scheme for configurations

   similar to Fig 145:  Geodesic dome,

   based on top 5 faces of icosahedron.

   See References 7 and 8. (*)

S=60;   (*) Span (*)

m=14;   (*) Frequency,

             m must be 5 plus a multiple of 3 (*)

p=(m-2)/3;

R=sqrt|(10+2*sqrt|5)*sin|36*S/4;

E=genid(m,m,2,1,1,-1)|{[0,0; 2,0],

   [0,0; 1,1], [2,0; 1,1]};

C=[0,0]#[2*m,0]#genid(p,p,6,3,3,-1)|

   [5,1]#genid(p+1,p+1,6,3,3,-1)|

   {[2,0], [2,2]};

G=pol(1,5,'[1-5]',R,[0,0; 2*m,0])|

   dil(2,sqrt|3)|lux(C)|E;

Dome=trac(1,[0,0,0,R],4,-15)|G;

use &,vm(2),vt(2),vh(S,S,3*S, 0,0,0, 0,0,1);

clear;   draw Dome;
برنامهء فرميينی برای تاشه های شبيه به شکل 146
(*) Formian scheme for configurations

   similar to Fig 146:  Geodesic dome of
   Fig 145, scaled vertically to make it
   ellipsoidal. See References 7 and 8. (*)

S=60;   (*) Span (*)

m=14;   (*) Frequency,

             m must be 5 plus a multiple of 3 (*)

f=1.75;    (*) Scale factor (*)

p=(m-2)/3;

R=sqrt|(10+2*sqrt|5)*sin|36*S/4;

E=genid(m,m,2,1,1,-1)|{[0,0; 2,0],

   [0,0; 1,1], [2,0; 1,1]};

C=[0,0]#[2*m,0]#genid(p,p,6,3,3,-1)|

   [5,1]#genid(p+1,p+1,6,3,3,-1)|

   {[2,0], [2,2]};

G=pol(1,5,'[1-5]',R,[0,0; 2*m,0])|

   dil(2,sqrt|3)|lux(C)|E;

Dome=dil(3,f)|trac(1,[0,0,0,R],4,-15)|G;

use &,vm(2),vt(2),

   vh(0,100000*S,0, 0,0,0, 0,0,1);

clear;   draw Dome;

برنامهء فرميينی برای تاشه های شبيه به شکل 147
(*) Formian scheme for configurations

   similar to Fig 147:  Geodesic dome of
   Fig 145, pellevated to add a conical
   effect. See References 7 and 8. (*)

S=60;    (*) Span (*)

H=12;   (*) Pellevational rise (*)

m=14;   (*) Frequency,

             m must be 5 plus a multiple of 3 (*)

p=(m-2)/3;

R=sqrt|(10+2*sqrt|5)*sin|36*S/4;

E=genid(m,m,2,1,1,-1)|{[0,0; 2,0],

   [0,0; 1,1], [2,0; 1,1]};

C=[0,0]#[2*m,0]#genid(p,p,6,3,3,-1)|[5,1]#

   genid(p+1,p+1,6,3,3,-1)|{[2,0], [2,2]};

G=pol(1,5,'[1-5]',R,[0,0; 2*m,0])|

   dil(2,sqrt|3)|lux(C)|E;

D=trac(1,[0,0,0,R],4,-15)|G;

Dome=capel(4,0,0,S,H)|D;

use &,vm(2),vt(2),

   vh(0,100000*S,0, 0,0,0, 0,0,1);

clear;   draw Dome;

برنامهء فرميينی برای تاشه های شبيه 

به شکلهای 148 و 149
(*) Formian scheme for configurations

   similar to Figs 148 (P=2) and 149
   (P=4):  Novated dome with binate

   pattern and edges held in position

   (novation mode 3 is used).

   See References 9, 10 and 11. (*)

L1=45;    (*) Length in U1 direction (*)

L2=45;    (*) Length in U2 direction (*)

m=18;     (*) Frequency in U1 direction (*)

n=18;      (*) Frequency in U2 direction (*)

Sh=11;    (*) Specifed height (*)

P=2;        (*) Position specifier (*)

Sb=[L1/P,L2/P,0];    (*) Specified base (*)

St=[L1/P,L2/P,Sh];   (*) Specified target (*)

E=rinid(m,n+1,1,1)|[0,0,0; 1,0,0]#

   rinid(m+1,n,1,1)|[0,0,0; 0,1,0];

Grid=bb(L1/m,L2/n)|E;

Edge=bb(L1/m,L2/n)|(rinid(m+1,2,1,n)|

   [0,0,0]#rinid(2,n-1,m,1)|[0,1,0]);

Dome=nov(3,1,Edge#Sb,Edge#St)|Grid;

use &,vm(2),vt(2),vh(L1/2,-L2,1.5*L2,

   L1/2,L2/2,0, L1/2,L2/2,1);

clear;   draw Dome;

برنامهء فرميينی برای تاشه های شبيه به شکل 150
(*) Formian scheme for configurations

   similar to Fig 150:  Novated dome with

   binate pattern and edges held in position

   (novation mode 2 is used).

   See References 9, 10 and 11. (*)

L1=45;    (*) Length in U1 direction (*)

L2=45;    (*) Length in U2 direction (*)

m=18;     (*) Frequency in U1 direction (*)

n=18;      (*) Frequency in U2 direction (*)

Sh=12;    (*) Specifed height (*)

Sb=[L1/2,L2/2,0];    (*) Specified base (*)

St=[L1/2,L2/2,Sh];   (*) Specified target (*)

E=rinid(m,n+1,1,1)|[0,0,0; 1,0,0]#

   rinid(m+1,n,1,1)|[0,0,0; 0,1,0];

Grid=bb(L1/m,L2/n)|E;

Edge=bb(L1/m,L2/n)|(rinid(m+1,2,1,n)|

   [0,0,0]#rinid(2,n-1,m,1)|[0,1,0]);

Dome=nov(2,1,Edge#Sb,Edge#St)|Grid;

use &,vm(2),vt(2),vh(L1/2,-L2,1.5*L2,

   L1/2,L2/2,0, L1/2,L2/2,1);

clear;   draw Dome;

برنامهء فرميينی برای تاشه های شبيه به شکل 151
(*) Formian scheme for configurations

   similar to Fig 151:  Novated dome

   with binate pattern, four raised points and

   edges held in position (novation mode 3 is

   used). See References 9, 10 and 11. (*)

L1=45;    (*) Length in U1 direction (*)

L2=45;    (*) Length in U2 direction (*)

m=18;     (*) Frequency in U1 direction (*)

n=18;      (*) Frequency in U2 direction (*)

Sh=5;      (*) Specifed height (*)

Sb=rinid(2,2,L1/2,L2/2)|[L1/4,L2/4,0];

St=proj(3,Sh)|Sb;

E=rinid(m,n+1,1,1)|[0,0,0; 1,0,0]#

   rinid(m+1,n,1,1)|[0,0,0; 0,1,0];

Grid=bb(L1/m,L2/n)|E;

Edge=bb(L1/m,L2/n)|(rinid(m+1,2,1,n)|

   [0,0,0]#rinid(2,n-1,m,1)|[0,1,0]);

Dome=nov(3,1,Edge#Sb,Edge#St)|Grid;

use &,vm(2),vt(2),vh(L1/2,-L2,1.5*L2,

   L1/2,L2/2,0, L1/2,L2/2,1);

clear;   draw Dome;

برنامهء فرميينی برای تاشه های شبيه 

به شکلهای 152، 153 و 154
(*) Formian scheme for configurations

   similar to Figs 152 (n=4), 153 (n=5)

   and 154 (n=6):  Sectorate novated dome

   with ternate pattern, a central raised point

   and with corners held in position.

   See References 9, 10 and 11. (*)

L=30;  (*) Length of each side (*)

m=7;    (*) Frequency (*)

n=4;     (*) Number of sides (*)

Sh=7;   (*) Specified height (*)

E=bb(L/(2*m),L/(2*m*tan|(180/n)))|

   genid(1,m,2,1,-1,1)|{[0,0,0; -1,1,0],

   [0,0,0; 1,1,0], [-1,1,0; 1,1,0]};

Grid=pex|rosad(0,0,n,360/n)|E;

Corner=rosad(0,0,n,360/n)|bb(L/(2*m),

   L/(2*m*tan|(180/n)))|[m,m,0];

Dome=nov(3,1,Corner#[0,0,0],

   Corner#[0,0,Sh])|Grid;

use &,vm(2),vt(2),vh(0,-2*L,1.5*L,

   0,0,0, 0,0,1);

clear;   draw Dome;

برنامهء فرميينی برای تاشه های شبيه به شکل 155
(*) Formian scheme for configurations

   similar to Fig 155:  Novated dome with

   ternate pattern and triangular boundary

   and with a central raised point and corners

   held in position.

   See References 9, 10 and 11. (*)

L=36;     (*) Length of each side (*)

m=12;    (*) Frequency (*)

Sh=L/7;  (*) Specifed height (*)

s=L*(sqrt|3)/2;

Corner=rosad(L/2,s/3,3,120)|[0,0,0];

Grid=bb(L/(2*m),s/m)|

   genid(m,m,2,1,1,-1)|{[0,0,0; 2,0,0],

   [1,1,0; 0,0,0], [1,1,0; 2,0,0]};

Dome=nov(3,1,Corner#[L/2,s/3,0],Corner#

   [L/2,s/3,Sh])|Grid;

use &,vm(2),vt(2),vh(L/2,-L,L,

   L/2,L/2,0, L/2,L/2,1);

clear;   draw Dome;

برنامهء فرميينی برای تاشه های شبيه به شکل 156
(*) Formian scheme for configurations

   similar to Fig 156:  Novated dome

   with binate pattern, a number of raised

   points and with edges held in position.

   See References 9, 10 and 11. (*)

L1=45;    (*) Length in U1 direction (*)

L2=45;    (*) Length in U2 direction (*)

m=27;     (*) Frequency in U1 direction,

                     must be a multiple of 3 (*)

n=27;      (*) Frequency in U2 direction,

                     must be a multiple of 3 (*)

m1=m/3;   n1=n/3;

Sh=3.5;   (*) Specifed height (*)

Sb=bb(L1/m,L2/n)|(rinid(2*m1+1,2,1,2*n1)|

   [m1/2,n1/2,0]#rinid(2,2*n1-1,2*m1,1)|

   [m1/2,1+n1/2,0]);

E=rinid(m,n+1,1,1)|[0,0,0; 1,0,0]#

   rinid(m+1,n,1,1)|[0,0,0; 0,1,0];

E=lux(rinid(m1-1,n1-1,1,1)|[m1+1,n1+1,0])|E;

Grid=bb(L1/m,L2/n)|E;

Edge=bb(L1/m,L2/n)|(rinid(m+1,2,1,n)|

   [0,0,0]#rinid(2,n-1,m,1)|[0,1,0]#

   rinid(m1+1,2,1,n1)|[m1,n1,0]#

   rinid(2,n1-1,m1,1)|[m1,n1+1,0]);

Dome=nov(3,1,Edge#Sb,Edge#

   proj(3,Sh)|Sb)|Grid;

use &,vm(2),vt(2),vh(L1/2,-L2,2*L2,

   L1/2,L2/2,0, L1/2,L2/2,1);

clear;   draw Dome;
برنامهء فرميينی برای تاشه های شبيه 

به شکلهای 157 و 158
(*) Formian scheme for configurations

   similar to Figs 157 (n=5) and 158
   (n=6):  Sectorate novated dome with

   ternate pattern, n raised points and with

   outside corners and inside edges held in

   position. See References 9, 10 and 11. (*)

L=30;   (*) Length of each side (*)

m=9;     (*) Frequency (*)

n=5;      (*) Number of sides (*)

Sh=2.5; (*) Specified height (*)

Sb=rosad(0,0,n,360/n)|bb(L/(2*m),

   L/(2*m*tan|(180/n)))|[m/2,m/2,0];

E=bb(L/(2*m),L/(2*m*tan|(180/n)))|

   lux({[0,0,0], [1,1,0], [-1,1,0]})|

  genid(1,m,2,1,-1,1)|{[0,0,0; -1,1,0],

   [0,0,0; 1,1,0], [-1,1,0; 1,1,0]};

Grid=pex|rosad(0,0,n,360/n)|E;

Corner=rosad(0,0,n,360/n)|bb(L/(2*m),

   L/(2*m*tan|(180/n)))|

   {[m,m,0], [2,2,0], [0,2,0]};

Dome=nov(3,1,Corner#Sb,

   Corner#proj(3,Sh)|Sb)|Grid;

use &,vm(2),vt(2),vh(0,-2*L,1.5*L,

   0,0,0, 0,0,1);

clear;   draw Dome;

برنامهء فرميينی برای تاشه های شبيه به شکل 159
(*) Formian scheme for configurations

   similar to Fig 159:  Novated dome

   with binate pattern, three raised points

   and edges held in position.

   See References 9, 10 and 11. (*)

L1=45;    (*) Length in U1 direction (*)

L2=45;    (*) Length in U2 direction (*)

m=24;     (*) Frequency in U1 direction,

                    must be a multiple of 3 (*)

n=24;      (*) Frequency in U2 direction,

                    must be a multiple of 3 (*)

Sh=6;      (*) Specifed height (*)

m1=m/3;  n1=n/3;  L11=L1/3;  L21=L2/3;

Sb={[L11,2*L21,0], [2*L11,2*L21,0],

   [2*L11,L21,0]};

E=rinid(m,n+1,1,1)|[0,0,0; 1,0,0]#

   rinid(m+1,n,1,1)|[0,0,0; 0,1,0];

E=lux(rinid(m1+1,n1+1,1,1)|[0,0,0])|E;

Grid=bb(L1/m,L2/n)|E;

Edge=bb(L1/m,L2/n)|

   (rinid(m1+1,2,1,2*n1-1)|

   [0,n1+1,0]#rinid(2*m1,2,1,n)|[m1+1,0,0]#

   rinid(2,n1+1,2*m1-1,1)|[m1+1,1,0]#

   rinid(2,2*m1-2,m,1)|[0,n1+2,0]);

Dome=nov(3,1,Edge#Sb,Edge#

   proj(3,Sh)|Sb)|Grid;

use &,vm(2),vt(2),vh(L1/2,-L2,1.5*L2,

   L1/2,L2/2,0, L1/2,L2/2,1);

clear;   draw Dome;

برنامهء فرميينی برای تاشه های شبيه به شکل 160
(*) Formian scheme for configurations

   similar to Fig 160:  Barrel-dome.

   See Section 3.5 of Reference 3. (*)

S=40; (*) Span of barrel vault

                             and the semi-domes (*)

H=12; (*) Rise of barrel vault

                             and the semi domes (*)

L=32; (*) Length of barrel vault (*)

m=6;  (*) Frequency of the semi-domes (*)

n=8;   (*) Frequency along the

     length of barrel vault, n must be even (*)

p=3;  (*) Number of sectors of

                                   each semi-dome (*)

A=2*atan|(2*H/S);     (*) Sweep angle (*)

R=S/(2*sin|A);           (*) Circumradius (*)

E=rinit(2*m+1,n,1,1)|[1,0,0; 1,0,1]#

   rinit(2*m,n-1,1,1)|[1,0,1; 1,1,1]#

   lamit(m,n/2)|rinit(m,n/2,1,1)|

   [1,1,0; 1,0,1];

B=verad(0,0,90-A)|bc(R,A/m,L/n)|E;

F=bd(R,180/p,A/m)|genit(1,m,1,1,0,1)|

   {[1,0,0; 1,0,1], [1,0,0; 1,1,1],

   [1,0,1; 1,1,1]};

D=pex|rosad(0,0,p,180/p)|F;

Domes=lam(2,-L/2)|D;

Barrel=verat(0,0)|B;

Bard=Barrel#Domes;

use &,vm(2),vt(2),

   vh(2*R,2*R,8*R, 0,0,0, 0,0,1);

clear;   draw Bard;
برنامهء فرميينی برای تاشه های شبيه به شکل 161
(*) Formian scheme for configurations

   similar to Fig 161:  Barrel-dome.

   See Section 3.5 of Reference 3. (*)

S=35; (*) Span of barrel vault

                             and the semi-domes (*)

H=10; (*) Rise of barrel vault

                             and the semi domes (*)

L=20; (*) Length of barrel vault (*)

m=6;  (*) Frequency of the semi-domes (*)

n=6;   (*) Frequency along the length of

                  barrel vault, n must be even (*)

p=3;  (*) Number of sectors of

                                  each semi-dome (*)

A=2*atan|(2*H/S);    (*) Sweep angle (*)

R=S/(2*sin|A);          (*) Circumradius (*)

E=rinit(2*m+1,n,1,1)|[1,0,0; 1,0,1]#

   rinit(2*m,n-1,1,1)|[1,0,1; 1,1,1]#

   lamit(m,n/2)|rinit(m,n/2,1,1)|

   [1,0,0; 1,1,1];

B=verad(0,0,90-A)|bc(R,A/m,L/n)|E;

F=bd(R,180/p,A/m)|genit(1,m,1,1,0,1)|

   {[1,0,0; 1,0,1], [1,0,0; 1,1,1],

   [1,0,1; 1,1,1]};

D=pex|rosad(0,0,p,180/p)|F;

Domes=lam(2,-L/2)|D;

Barrel=verat(0,0)|B;

Bard=Barrel#Domes;

use &,vm(2),vt(2),

   vh(2*R,2*R,8*R, 0,0,0, 0,0,1);

clear;   draw Bard;
برنامهء فرميينی برای تاشه های شبيه به شکل 162
(*) Formian scheme for configurations

   similar to Fig 162:  Barrel-dome.

   See Section 3.5 of Reference 3. (*)

S=35; (*) Span of barrel vault

                              and the semi-domes (*)

H=10; (*) Rise of barrel vault

                               and the semi domes (*)

L=25;  (*) Length of barrel vault (*)

m=6;   (*) Frequency of the semi-domes (*)

n=8;    (*) Frequency along the length

                of barrel vault, n must be even (*)

p=3;    (*) Number of sectors of

                                    each semi-dome (*)

A=2*atan|(2*H/S);      (*) Sweep angle (*)

R=S/(2*sin|A);            (*) Circumradius (*)

E=rinit(2*m+1,n,1,1)|[1,0,0; 1,0,1]#

   rinit(2*m,n-1,1,1)|[1,0,1; 1,1,1]#

   lamit(m,n/2)|rinit(m,n/2,1,2)|

   lam(3,1)|[1,1,0; 1,0,1];

E=lux(rin(3,n/2,2)|[1,m,1])|E;

B=verad(0,0,90-A)|bc(R,A/m,L/n)|E;

F=bd(R,180/p,A/m)|genit(1,m,1,1,0,1)|

   {[1,0,0; 1,0,1], [1,0,0; 1,1,1],

   [1,0,1; 1,1,1]};

D=pex|rosad(0,0,p,180/p)|F;

Domes=lam(2,-L/2)|D;

Barrel=verat(0,0)|B;

Bard=Barrel#Domes;

use &,vm(2),vt(2),

   vh(2*R,2*R,8*R, 0,0,0, 0,0,1);

clear;   draw Bard;
برنامهء فرميينی برای تاشه های شبيه به شکل 163
(*) Formian scheme for configurations

   similar to Fig 163:  Barrel-dome.

   See Section 3.5 of Reference 3. (*)

S=35;  (*) Span of barrel vault

                              and the semi-domes (*)

H=10;  (*) Rise of barrel vault

                               and the semi domes (*)

T=45;  (*) Central angle

                      containing the barrel vault (*)

R=60;  (*) Radius of annulus (*)

f=5;     (*) Frequency of the semi-domes (*)

m=10;  (*) Frequency along the length

                of barrel vault, m must be even (*)

p=3;    (*) Number of sectors of

                                    each semi-dome (*)

A=2*atan|(2*H/S);      (*) Sweep angle (*)

Q=S/(2*sin|A);            (*) Radius of

                        cross-section of annulus (*)

E=rinic(2,4,m,2*f+1,1,1)|[1,0,1,-f;

   1,1,1,-f]#rinic(2,4,m+1,2*f,1,1)|

   [1,0,1,-f; 1,0,1,1-f]#

   lamic(2,4,ric|(m/2),0)|rinic(2,4,ric|

   (m/2),f,1,1)|[1,0,1,0; 1,1,1,1];

Barrel=dep(4)|ba(R,T/m,Q,A/f)|E;

F=bd(Q,180/p,A/f)|genit(1,f,1,1,0,1)|

   {[1,0,0; 1,0,1], [1,0,0; 1,1,1],

   [1,0,1; 1,1,1]};

D=tran(1,R)|rosad(0,0,p,180/p)|F;

Domes=ref(2,0)|D#verad(0,0,T)|D;

Bard=pex|(Barrel#Domes);

use &,vm(2),vt(2),

   vh(2*S,4*S,6*S, 0,0,0, 0,0,1);

clear;   draw Bard;
برنامهء فرميينی برای تاشه های شبيه به شکل 164
(*) Formian scheme for configurations

   similar to Fig 164:  Barrel-dome.

   See Section 3.5 of Reference 3. (*)

S=35;  (*) Span of barrel vault

                              and the semi-domes (*)

H=10;  (*) Rise of barrel vault

                               and the semi domes (*)

T=90;  (*) Central angle

                      containing the barrel vault (*)

R=60; (*) Radius of annulus (*)

f=5;     (*) Frequency of the semi-domes (*)

m=20;  (*) Frequency along the length

                of barrel vault, m must be even (*)

p=3;    (*) Number of sectors of

                                    each semi-dome (*)

A=2*atan|(2*H/S);      (*) Sweep angle (*)

Q=S/(2*sin|A);            (*) Radius of

                         cross-section of annulus (*)

E=rinic(2,4,m,2*f+1,1,1)|[1,0,1,-f;

   1,1,1,-f]#rinic(2,4,m+1,2*f,1,1)|

   [1,0,1,-f; 1,0,1,1-f]#

   lamic(2,4,ric|(m/2),0)|rinic(2,4,ric|

   (m/2),f,1,1)|[1,1,1,0; 1,0,1,1];

Barrel=dep(4)|ba(R,T/m,Q,A/f)|E;

F=bd(Q,180/p,A/f)|genit(1,f,1,1,0,1)|

   {[1,0,0; 1,0,1], [1,0,0; 1,1,1],

   [1,0,1; 1,1,1]};

D=tran(1,R)|rosad(0,0,p,180/p)|F;

Domes=ref(2,0)|D#verad(0,0,T)|D;

Bard=pex|(Barrel#Domes);

use &,vm(2),vt(2),

   vh(0,-2*S,8*S, 0,0,0, 0,0,1);

clear;   draw Bard;
برنامهء فرميينی برای تاشه های شبيه به شکل 165
(*) Formian scheme for configurations

   similar to Fig 165:  Barrel-dome.

   See Section 3.5 of Reference 3. (*)

S=35;  (*) Span of barrel vault

                              and the semi-domes (*)

H=10;  (*) Rise of barrel vault

                               and the semi domes (*)

T=25;  (*) Central angle

                containing half the barrel vault (*)

R=60; (*) Radius of annulus (*)

f=5;    (*) Frequency of the semi-domes (*)

m=6;  (*) Frequency along

                  half the length of barrel vault (*)

p=3;   (*) Number of sectors of

                                   each semi-dome (*)

A=2*atan|(2*H/S);      (*) Sweep angle (*)

Q=S/(2*sin|A);            (*) Radius of

                        cross-section of annulus (*)

E=rinic(2,4,m,2*f+1,1,1)|[1,0,1,-f;

   1,1,1,-f]#rinic(2,4,m+1,2*f,1,1)|

   [1,0,1,-f; 1,0,1,1-f]#

   lam(4,0)|rinic(2,4,m,f,1,1)|

   [1,0,1,0; 1,1,1,1];

B=dep(4)|ba(R,T/m,Q,A/f)|E;

F=bd(Q,180/p,A/f)|genit(1,f,1,1,0,1)|

   {[1,0,0; 1,0,1], [1,0,0; 1,1,1],

   [1,0,1; 1,1,1]};

D=tran(1,R)|rosad(0,0,p,180/p)|F;

BD=B#verad(0,0,T)|D;

Bard=pex|(BD#verad(R,0,180)|BD);

use &,vm(2),vt(2),

   vh(3*S,4*S,8*S, 0,0,0, 0,0,1);

clear;   draw Bard;

برنامهء فرميينی برای تاشه های شبيه 

به شکلهای 166 و 167
(*) Formian scheme for configurations

   similar to Fig 166 (H=3) and Fig 167
   (H=12):  Simple hypar.  See Section 3.2

   of Reference 3. (*)

L=20;     (*) Length of side in plan (*)

H=3;       (*) Height (*)

m=6;       (*) Semi-frequency (*)

P=H/(L*L);     (*) Pitch (*)

n=2*m;

E=rinid(n,n+1,1,1)|[-m,-m,0; 1-m,-m,0]#

   rinid(n+1,n,1,1)|[-m,-m,0; -m,1-m,0];

Hypar=bhp(L/n,L/n,1,P)|E;

use &,vm(2),vt(2),vh(L,-L,L,0,0,0, 0,0,1);

clear;   draw Hypar;
برنامهء فرميينی برای تاشه های شبيه به شکل 168
(*) Formian scheme for configurations

   similar to Fig 168:  Double layer hypar.

   See Section 3.2 of Reference 3. (*)

L=30;       (*) Length of side

                              of top layer in plan (*)

D=1.5;     (*) Distance between layers (*)

m=5;        (*) Frequency of top layer (*)

P=0.015;  (*) Pitch (*)

T=pex|rinid(m,m,2,2)|rosad(1-m,1-m)|

   [-m,-m,1; 2-m,-m,1];

B=pex|rinid(m-1,m-1,2,2)|rosad(2-m,2-m)|

   [1-m,1-m,0; 3-m,1-m,0];

W=rinid(m,m,2,2)|rosad(1-m,1-m)|

   [-m,-m,1; 1-m,1-m,0];

Hypar=bhp(L/(2*m),L/(2*m),D,P)|

   (T#B#W);

use &,vm(2),vt(2),vh(L,-L,2*L,0,0,0, 0,0,1);

clear;   draw Hypar;

برنامهء فرميينی برای تاشه های شبيه به شکل 169
(*) Formian scheme for configurations

   similar to Fig 169:  Compound hypar

   (convex).  See Section 3.2 of

   Reference 3. (*)

L=20;     (*) Length of side

                     of each hypar units in plan (*)

H=7;       (*) Height (*)

m=4;       (*) Semi-frequency (*)

P=H/(L*L);     (*) Pitch (*)

n=2*m;

E=rinid(n,n+1,1,1)|[-m,-m,0; 1-m,-m,0]#

   rinid(n+1,n,1,1)|[-m,-m,0; -m,1-m,0];

F=bhp(L/n,L/n,1,P)|E;

Hypar=pex|rosad(L/2,L/2)|F;

use &,vm(2),vt(2),

   vh(0.1*L,-4*L,4*L, 0,0,0, 0,0,1);

clear;   draw Hypar;
برنامهء فرميينی برای تاشه های شبيه به شکل 170
(*) Formian scheme for configurations

   similar to Fig 170:  Compound hypar

   (concave).  See Section 3.2 of

   Reference 3. (*)

L=20;     (*) Length of side

                     of each hypar units in plan (*)

H=7;      (*) Height (*)

m=4;      (*) Semi-frequency (*)

P=H/(L*L);     (*) Pitch (*)

n=2*m;

E=rinid(n,n+1,1,1)|[-m,-m,0; 1-m,-m,0]#

   rinid(n+1,n,1,1)|[-m,-m,0; -m,1-m,0];

F=bhp(L/n,L/n,1,P)|E;

Hypar=pex|rosad(L/2,-L/2)|F;

use &,vm(2),vt(2),

   vh(0.1*L,-3*L,3*L, 0,0,0, 0,0,1);

clear;   draw Hypar;
برنامهء فرميينی برای تاشه های شبيه به شکل 171
(*) Formian scheme for configurations

   similar to Fig 171:  Compound hypar.

   See Section 3.2 of Reference 3. (*)

L=20;     (*) Length of side

                     of each hypar units in plan (*)

H=8;       (*) Height (*)

m=3;       (*) Semi-frequency (*)

P=H/(L*L);     (*) Pitch (*)

n=2*m;

E=rinid(n,n+1,1,1)|[-m,-m,0; 1-m,-m,0]#

   rinid(n+1,n,1,1)|[-m,-m,0; -m,1-m,0];

F=bhp(L/n,L/n,1,P)|E;

Hypar=pex|rosad(3*L/2,3*L/2)|

   rosad(L/2,L/2)|F;

use &,vm(2),vt(2),

   vh(6*L,6*L,5*L,0,0,0, 0,0,1);

clear;   draw Hypar;
برنامهء فرميينی برای تاشه های شبيه به شکل 172
(*) Formian scheme for configurations

   similar to Fig 172:  Pellevated compound

   hypar. See Section 3.2 of Reference 3

   and References 4 and 5. (*)

L=20;     (*) Length of side

                       of each hypar units in plan (*)

H=7;       (*) Original height (*)

R=7;       (*) Pellevational rise (*)

m=4;       (*) Semi-frequency (*)

P=H/(L*L);     (*) Pitch (*)

n=2*m;

E=rinid(n,n+1,1,1)|[-m,-m,0; 1-m,-m,0]#

   rinid(n+1,n,1,1)|[-m,-m,0; -m,1-m,0];

F=bhp(L/n,L/n,1,P)|E;

G=pex|rosad(L/2,L/2)|F;

Hypar=capel(3,L/2,L/2,2*L*sqrt|2,R)|G;

use &,vm(2),vt(2),

   vh(0.1*L,-3*L,3*L, 0,0,0, 0,0,1);

clear;   draw Hypar;
برنامهء فرميينی برای تاشه های شبيه به شکل 173
(*) Formian scheme for configurations

   similar to Fig 173 :  Pellevated compound

   hypar. See Section 3.2 of Reference 3

   and References 4, 5 and 6. (*)

L=20;     (*) Length of side

                     of each hypar units in plan (*)

H=8;       (*) Original height (*)

R=35;     (*) Pellevational rise (*)

m=3;       (*) Semi-frequency (*)

P=H/(L*L);     (*) Pitch (*)

n=2*m;

E=rinid(n,n+1,1,1)|[-m,-m,0; 1-m,-m,0]#

   rinid(n+1,n,1,1)|[-m,-m,0; -m,1-m,0];

F=bhp(L/n,L/n,1,P)|E;

G=pex|rosad(3*L/2,3*L/2)|

   rosad(L/2,L/2)|F;

Hypar=capel(3,3*L/2,3*L/2,

   4*L*sqrt|2,R)|G;

use &,vm(2),vt(2),

   vh(20*L,20*L,25*L,0,0,0, 0,0,1);

clear;   draw Hypar;

برنامهء فرميينی برای تاشه های شبيه به شکل 174
(*) Formian scheme for configurations

   similar to Fig 174:  Pyramid.

   See Section 2.2 of Reference 2. (*)

L=20; (*) Length of each side of base (*)

H=20; (*) Height of pyramid (*)

m=7;   (*) Frequency (*)

n=3;    (*) Number of sides of the base (*)

C=L/(2*tan|(180/n));

S=sqrt|(C*C+H*H);

B=atan|(H/C);

E1=genid(1,m,2,1,-1,1)|{[0,0,0; -1,1,0],

   [0,0,0; 1,1,0], [-1,1,0; 1,1,0]};

E2=bb(L/(2*m),S/m)|E1;

E3=tran(2,C-S)|E2;

E4=verat(C,0,-B)|E3;

Pyramid=pex|rosad(0,0,n,360/n)|E4;

use &,vm(2),vt(2),

   vh(0,-L,9*L, 0,0,0, 0,0,1);

clear;   draw Pyramid;
برنامهء فرميينی برای تاشه های شبيه 
به شکلهای 175 و 176
(*) Formian scheme for configurations

   similar to Fig 175 (m=9, n=5) and

   Fig 176 (m=6, n=12):  Pyramid.

   See Section 2.2 of Reference 2. (*)

L=20; (*) Length of each side of base (*)

H=15; (*) Height of pyramid (*)

m=9;   (*) Frequency (*)

n=5;    (*) Number of sides of the base (*)

C=L/(2*tan|(180/n));

S=sqrt|(C*C+H*H);

B=atan|(H/C);

E1=genid(1,m,2,1,-1,1)|{[0,0,0; -1,1,0],

   [0,0,0; 1,1,0], [-1,1,0; 1,1,0]};

E2=bb(L/(2*m),S/m)|E1;

E3=tran(2,C-S)|E2;

E4=verat(C,0,-B)|E3;

Pyramid=pex|rosad(0,0,n,360/n)|E4;

use &,vm(2),vt(2),

   vh(0,-L,3*L, 0,0,0, 0,0,1);

clear;   draw Pyramid;
برنامهء فرميينی برای تاشه های شبيه به شکل 177
(*) Formian scheme for configurations

   similar to Fig 177:  Pyramid.

   See Section 2.2 of Reference 2. (*)

L=20; (*) Length of each side of base (*)

H=20; (*) Height of pyramid (*)

m=8;   (*) Frequency, m must be even (*)

n=9;    (*) Number of sides of the base (*)

C=L/(2*tan|(180/n));

S=sqrt|(C*C+H*H);

B=atan|(H/C);

mm=ric|(m/2)-1;

E=genid(1,m,2,1,-1,1)|{[0,0,0; -1,1,0],

   [0,0,0; 1,1,0], [-1,1,0; 1,1,0]};

E1=lux(genid(1,3,2,1,-1,1)|[0,m-2,0])|E;

E2=bb(L/(2*m),S/m)|E1;

E3=tran(2,C-S)|E2;

E4=verat(C,0,-B)|E3;

Pyramid=pex|rosad(0,0,n,360/n)|E4;

use &,vm(2),vt(2),

   vh(L,L,2.5*L, 0,0,0, 0,0,1);

clear;   draw Pyramid;
برنامهء فرميينی برای تاشه های شبيه به شکل 178
(*) Formian scheme for configurations

   similar to Fig 178:  Pyramid.

   See Section 2.2 of Reference 2. (*)

L=20; (*) Length of each side of base (*)

H=20; (*) Height of pyramid (*)

m=15; (*) Frequency (*)

n=6;    (*) Number of sides of the base (*)

C=L/(2*tan|(180/n));

S=sqrt|(C*C+H*H);

B=atan|(H/C);

mm=ric|(m/2)-1;

E=genid(1,m,2,1,-1,1)|{[0,0,0; -1,1,0],

   [0,0,0; 1,1,0], [-1,1,0; 1,1,0]};

E1=lux(genid(1,5,4,2,-2,1)|[0,4,0])|E;

E2=bb(L/(2*m),S/m)|E1;

E3=tran(2,C-S)|E2;

E4=verat(C,0,-B)|E3;

Pyramid=pex|rosad(0,0,n,360/n)|E4;

use &,vm(2),vt(2),

   vh(0,2*L,5*L, 0,0,0, 0,0,1);

clear;   draw Pyramid;
برنامهء فرميينی برای تاشه های شبيه به شکل 179
(*) Formian scheme for configurations

   similar to Fig 179:  Double layer pyramid.

   See Section 2.2 of Reference 2. (*)

L=20;     (*) Length of each side of base (*)

H=12;    (*) Height of pyramid (*)

D=1.25; (*) Distance between two layers (*)

m=4;      (*) Frequency (*)

n=6;       (*) Number of sides of the base (*)

C=L/(2*tan|(180/n));

S=sqrt|(C*C+H*H);

B=atan|(H/C);

Et=genid(1,m,2,3,-1,1)|{[0,0,0; -1,3,0],

   [0,0,0; 1,3,0], [-1,3,0; 1,3,0]};

Eb=genid(1,m-1,2,3,-1,1)|{[0,2,-1; -1,5,-1],

   [0,2,-1; 1,5,-1], [-1,5,-1; 1,5,-1]};

Ew=genid(1,m,2,3,-1,1)|{[0,0,0; 0,2,-1],

   [-1,3,0; 0,2,-1], [1,3,0; 0,2,-1]};

E1=Et#Eb#Ew;

E2=bt(L/(2*m),S/(3*m),D)|E1;

E3=tran(2,C-S)|E2;

E4=verat(C,0,-B)|E3;

Qi=[0,2,-1];

Ri=verat(C,0,-B)|tran(2,C-S)|

   bt(L/(2*m),S/(3*m),D)|Qi;

Rj=verad(0,0,360/n)|Ri;

K1=tig|(Ri#Rj);

K2=rinax(0,0,H, -L/(2*m),

   C/m,(m-1)*H/m, m)|K1;

(*) K2 represents the bottom layer

     elements that link the n segments

     of the pyramid (*)

Pyramid=pex|rosad(0,0,n,360/n)|(E4#K2);

use &,vm(2),vt(2),

   vh(0,2*L,3*L, 0,0,0, 0,0,1);

clear;   draw Pyramid;
برنامهء فرميينی برای تاشه های شبيه به شکل 180
(*) Formian scheme for configurations

   similar to Fig 180 :  Pellevated pyramid.

   See Section 2.2 of Reference 2 and

   References 4, 5 and 6. (*)

L=20; (*) Length of each side of base (*)

H=18; (*) Height of pyramid (*)

P=8;   (*) Pellevational rise (*)

m=9;   (*) Frequency (*)

n=5;    (*) Number of sides of the base (*)

C=L/(2*tan|(180/n));

S=sqrt|(C*C+H*H);

B=atan|(H/C);

E1=genid(1,m,2,1,-1,1)|{[0,0,0; -1,1,0],

   [0,0,0; 1,1,0], [-1,1,0; 1,1,0]};

E2=bb(L/(2*m),S/m)|E1;

E3=tran(2,C-S)|E2;

E4=verat(C,0,-B)|E3;

E5=pex|rosad(0,0,n,360/n)|E4;

Pyramid=capel(3,0,0,

   L/sin|(180/n),P)|E5;

use &,vm(2),vt(2),

   vh(0,-L,4.5*L, 0,0,0, 0,0,1);

clear;   draw Pyramid;
برنامهء فرميينی برای تاشه های شبيه به شکل 181
(*) Formian scheme for configurations

   similar to Fig 181:  Pellevated Pyramid.

   See Section 2.2 of Reference 2

   and References 4, 5 and 6. (*)

L=20; (*) Length of each side of base (*)

H=20; (*) Height of pyramid (*)

P=6;   (*) Pellevational rise (*)

m=15; (*) Frequency (*)

n=6;    (*) Number of sides of the base (*)

C=L/(2*tan|(180/n));

S=sqrt|(C*C+H*H);

B=atan|(H/C);

mm=ric|(m/2)-1;

E=genid(1,m,2,1,-1,1)|{[0,0,0; -1,1,0],

   [0,0,0; 1,1,0], [-1,1,0; 1,1,0]};

E1=lux(genid(1,5,4,2,-2,1)|[0,4,0])|E;

E2=bb(L/(2*m),S/m)|E1;

E3=tran(2,C-S)|E2;

E4=verat(C,0,-B)|E3;

E5=pex|rosad(0,0,n,360/n)|E4;

Pyramid=capel(3,0,0,

   L/sin|(180/n),P)|E5;

use &,vm(2),vt(2),

   vh(0,2*L,5.5*L, 0,0,0, 0,0,1);

clear;   draw Pyramid;
برنامهء فرميينی برای تاشه های شبيه به شکل 182
(*) Formian scheme for configurations

   similar to Fig 182:  Pellevated pyramid.

   See Section 2.2 of Reference 2 and

   References 4, 5 and 6. (*)

L=20; (*) Length of each side of base (*)

H=18; (*) Height of pyramid (*)

P=8;   (*) Pellevational rise (*)

m=9;   (*) Frequency (*)

n=5;    (*) Number of sides of the base (*)

C=L/(2*tan|(180/n));

S=sqrt|(C*C+H*H);

B=atan|(H/C);

E1=genid(1,m,2,1,-1,1)|{[0,0,0; -1,1,0],

   [0,0,0; 1,1,0], [-1,1,0; 1,1,0]};

E2=bb(L/(2*m),S/m)|E1;

E3=tran(2,C-S)|E2;

E4=verat(C,0,-B)|E3;

E5=pex|rosad(0,0,n,360/n)|E4;

Pyramid=capel(-3,0,0,

   L/sin|(180/n),P)|E5;

use &,vm(2),vt(2),

   vh(0,-L,2.5*L, 0,0,0, 0,0,1);

clear;   draw Pyramid;

برنامهء فرميينی برای تاشه های شبيه به شکل 183
(*) Formian scheme for configurations

   similar to Fig 183:  Pellevated Pyramid.

   See Section 2.2 of Reference 2

   and References 4, 5 and 6. (*)

L=20; (*) Length of each side of base (*)

H=20; (*) Height of pyramid (*)

P=6;   (*) Pellevational rise (*)

m=15; (*) Frequency (*)

n=6;    (*) Number of sides of the base (*)

C=L/(2*tan|(180/n));

S=sqrt|(C*C+H*H);

B=atan|(H/C);

mm=ric|(m/2)-1;

E=genid(1,m,2,1,-1,1)|{[0,0,0; -1,1,0],

   [0,0,0; 1,1,0], [-1,1,0; 1,1,0]};

E1=lux(genid(1,5,4,2,-2,1)|[0,4,0])|E;

E2=bb(L/(2*m),S/m)|E1;

E3=tran(2,C-S)|E2;

E4=verat(C,0,-B)|E3;

E5=pex|rosad(0,0,n,360/n)|E4;

Pyramid=capel(-3,0,0,

   L/sin|(180/n),P)|E5;

use &,vm(2),vt(2),

   vh(0,2*L,3.5*L, 0,0,0, 0,0,1);

clear;   draw Pyramid;
برنامهء فرميينی برای تاشه های شبيه به شکل 184
(*) Formian scheme for configurations

   similar to Fig 184:  Freeform (novated

   single layer grid with binate pattern and

   with corners held in position (novation

   mode 3 is used)). See References 9, 10

   and 11. (*)

Grid=rinid(24,25,1,1)|[0,0,0; 1,0,0]#

   rinid(25,24,1,1)|[0,0,0; 0,1,0];

Corner=lamid(12,12)|[0,0,0];

Base={[14,2,0], [8,18,0], [20,16,0],

   [4,22,0]};

Target={[14,2,5], [8,18,0], [20,16,3],

   [4,22,2]};

Form=nov(3,1,Corner#Base,

   Corner#Target)|Grid;

use &,vm(2),vt(2),vh(12,-24,24,

   12,12,0, 12,12,1);

clear;   draw Form;

برنامهء فرميينی برای تاشه های شبيه به شکل 185
(*) Formian scheme for configurations

   similar to Fig 185:  Freeform (pellevated

   single layer grid with binate pattern).

   See References 4, 5 and 6. (*)

Grid=rinid(40,41,1,1)|[0,0,0; 1,0,0]#

   rinid(41,40,1,1)|[0,0,0; 0,1,0];

Form=capel(3,12,16,20,4)|

   capel(3,18,16,24,5)|

   capel(3,24,20,28,6)|Grid;

Form=rel(U(1,3)>0 || U(2,3)>0)|Form;

use &,vm(2),vt(2),vh(20,-40,120,

   20,20,0, 20,20,1);

clear;   draw Form;

برنامهء فرميينی برای تاشه های شبيه به شکل 186
(*) Formian scheme for configurations

   similar to Fig 186:  Freeform (novated

   diamatic dome with nodes along 2/3 of the

   perimeter held in position (novation mode

   2 is used)). See Section 2.5 of Reference 2

   and References 9, 10 and 11. (*)

R=36;        (*) Radius of circumsphere (*)

A=40;        (*) Sweep angle (*)

m=8;          (*) Frequency (*)

n=6;           (*) Number of sectors (*)

E=genit(1,m,1,1,0,1)|{[1,0,0; 1,0,1],

   [1,0,0; 1,1,1], [1,0,1; 1,1,1]};

F=bd(R,360/n,A/m)|E;

Dome=pex|rosad(0,0,n,360/n)|F;

Edge=bd(R,360/n,A/m)|rin(2,33,-1)|[1,m,m];

Sb=bd(R,360/n,A/m)|[1,2*m,m];

Form=nov(2,1,Edge#Sb,Edge#

   tran(3,R/2)|Sb)|Dome;

use &,vm(2),vt(2),vh(R,R,3.5*R,

   0,0,0, 0,0,1);

clear;   draw Form;

برنامهء فرميينی برای تاشه های شبيه به شکل 187
(*) Formian scheme for configurations

   similar to Fig 187:  Freeform (pellevated

   diamatic dome). See Section 2.5 of

   Reference 2 and References 4, 5 and 6 (*)

R=36;      (*) Radius of circumsphere (*)

A=45;      (*) Sweep angle (*)

m=9;        (*) Frequency (*)

n=7;         (*) Number of sectors (*)

E=genit(1,m,3,3,0,1)|{[1,2,4; 1,1,2],

   [1,2,4; 1,1,5], [1,2,4; 1,4,5]}#

   rin(3,m+1,3)|[1,1,2; 1,-1,2];

F=bd(R,360/n,A/(3*m))|E;

Dome=rosad(0,0,n,360/n)|F;

Form=verad(0,0,-45)|bapel(3,R/2,0,R/2,R/2,

   0.7*R,0.2*R)|bapel(3,-R/2,0,-R/2,R/2,

   0.7*R,0.2*R)|Dome;

use &,vm(2),vt(2),vh(R,R,3*R, 0,0,0, 0,0,1);

clear;   draw Form;

برنامهء فرميينی برای تاشه های شبيه به شکل 188
(*) Formian scheme for configurations

   similar to Fig 188:  Freeform (pellevated

   compound hypar). See Section 3.2 of

   Reference 3 and References 4, 5 and 6. (*)

L=20;   (*) Length of side

                  of each hypar units in plan (*)

H=24;   (*) Height of each unit(*)

m=8;     (*) Semi-frequency of each unit(*)

P=H/(L*L);     (*) Pitch (*)

n=2*m;

E=rinid(n,n+1,1,1)|[-m,-m,0; 1-m,-m,0]#

   rinid(n+1,n,1,1)|[-m,-m,0; -m,1-m,0];

F=bhp(L/n,L/n,1,P)|E;

Hypar=pex|rosad(L/2,L/2)|F;

Form=bapel(3,0,0,0,L,L,L/2)|

   bapel(3,L,0,L,L,L,L/2)|Hypar;

use &,vm(2),vt(2),

   vh(0,-3*L,6*L, 0,0,0, 0,0,1);

clear;   draw Form;

برنامهء فرميينی برای تاشه های شبيه به شکل 189
(*) Formian scheme for configurations

   similar to Fig 189:  Freeform (pellevated

   binate grid). See References 4, 5 and 6. (*)

L=40;   (*) Length of each side of grid (*)

R=8;     (*) Rise of pellevants (*)

m=24;   (*) Frequency of grid (*)

A=L/2;

Grid=bb(L/m,L/m)|
   (rinid(m,m+1,1,1)|[0,0,0; 1,0,0]#

   rinid(m+1,m,1,1)|[0,0,0; 0,1,0]);

Form=capel(3,0,0,L,R)|capel(3,0,L,L,R)|
   capel(3,L,0,L,R)|capel(3,L,L,L,R)|
   capel(1,A,A,L*sqrt|2,R)|Grid;
use &,vm(2),vt(2),

   vh(A,-L,1.5*L, A,A,0, A,A,1);

clear;   draw Form;

برنامهء فرميينی برای تاشه های شبيه به شکل 190
(*) Formian scheme for configurations

   similar to Fig 190:  Freeform (pellevated

   geodesic dome). See References
   4, 5, 6, 7 and 8. (*)

RC=50; (*) Radius of circumsphere (*)

SP=30;  (*) Span of cylindrical pellevant (*)
Rise=7;  (*) Rise of pellevant (*)

m=9;      (*) Frequency of dome (*)

t=sqrt|3;

D=pol(1,5,’[1-5]’,RC,[0,0; 2*m,0])|

   genid(m,m,2,t,1,-1)|rosad(1,t/3,3,120)|

   [0,0; 2,0];
Geodome=dil(3,1/3)|

   trac(1,[0,0,0,RC],4,-15)|D;

Form=bapel(3,0,0,0,10,SP,Rise)|Geodome;
use &,vm(2),vt(2),

   vh(RC,RC,2*RC, 0,0,0, 0,0,1);

clear;   draw Form;

برنامهء فرميينی برای تاشه های شبيه به شکل 191
(*) Formian scheme for configurations

   similar to Fig 191:  Freeform (pellevated
   ring barrel vault with binate pattern).

   See References 4, 5 and 6 and Section 3.4
   of Reference 3. (*)

S=60;    (*) Span of barrel vault (*)

SP=80;  (*) Span of cylindrical pellevants (*)

H=14;   (*) Rise of barrel vault (*)

RP=16; (*) Rise of pellevants (*)

m=70;   (*) Frequency in U2 direction (*)

n=10;    (*) Frequency in U4 direction (*)

R=70;   (*) Radius of annulus (*)

A=2*atan|(2*H/S);    (*) Sweep angle (*)

Q=S/(2*sin|A);          (*) Radius of

                      cross-section of annulus (*)

E=rinic(2,4,m,n+1,1,1)|

   [1,0,1,-n/2; 1,1,1,-n/2]#

   rinic(2,4,m+1,n,1,1)|

   [1,0,1,-n/2; 1,0,1,1-n/2];

F=dep(4)|ba(R,360/m,Q,2*A/n)|E;

Form=bapel(3,0,0,R,0,SP,RP)|

   bapel(3,0,0,0,R,SP,RP)|F;

use &,vm(2),vt(2),

   vh(2*R,2*R,4*R, 0,0,0, 0,0,1);

clear;   draw Form;

برنامهء فرميينی برای تاشه های شبيه به شکل 192
(*) Formian scheme for configurations

   similar to Fig 192:  Freeform (paragenic

   configuration). See Section 3.6 of

   Reference 3. (*)

m=25;   (*) Frequency (*)

E=rinit(m,m+1,1,1)|[1,0,0; 1,1,0]#

   rinit(m+1,m,1,1)|[1,0,0; 1,0,1];

Form=bhp(1,1,1,0.01)|verad(0,0,-45)|

   tran(2,25)|verat(0,0)|bc(1,7.5,1)|

   verad(0,0,45)|bc(30,80/m,60/m)|E;

use &,vm(2),vt(2),

   vh(30,30,60, 0,0,30, 0,1,30);

clear;   draw Form;

برنامهء فرميينی برای تاشه های شبيه به شکل 193
(*) Formian scheme for configurations

   similar to Fig 193:  Hyperboloidal tower.

   See Section 3.3 of Reference 3. (*)

H=30;   (*) Height of tower (*)

H1=20; (*) Height at the neck of tower (*)

R=8;     (*) Radius at the base of tower (*)

R1=3;   (*) Radius at the neck of tower (*)

m=10;   (*) Number of sides of the base (*)

n=5;      (*) Number of rhombic units in

                               the vertical direction (*)

B=-2*n*H1/H;

P=(R*R-R1*R1)/(H1*H1);

E=rinit(m,n,2,2)|(lamit(0,B+1)|

  [1,0,B; 1,-1,B+1]#{[1,-1,B+1; 1,1,B+1],

  [1,0,B+2; 1,2,B+2]});

Tower=bh(R1,360/(2*m),H/(2*n),P)|E;

use &,vm(2),vt(2),

   vh(10*R,10*R,-R, 0,0,0, 0,0,1);

clear;   draw Tower;
برنامهء فرميينی برای تاشه های شبيه به شکل 194
(*) Formian scheme for configurations

   similar to Fig 194:  Hyperboloidal tower.

   See Section 3.3 of Reference 3. (*)

H=60;   (*) Height of tower (*)

H1=50; (*) Height at the neck of tower (*)

R=8;     (*) Radius at the base of tower (*)

R1=3;   (*) Radius at the neck of tower (*)

m=6;     (*) Number of sides of the base (*)

n=8;      (*) Number of rhombic units in

                               the vertical direction (*)

B=-2*n*H1/H;

P=(R*R-R1*R1)/(H1*H1);

E=rinit(m,n,2,2)|(lamit(0,B+1)|

  [1,0,B; 1,-1,B+1]#{[1,-1,B+1; 1,1,B+1],

  [1,0,B+2; 1,2,B+2]});

Tower=bh(R1,360/(2*m),H/(2*n),P)|E;

use &,vm(2),vt(2),

   vh(10*R,10*R,-4*R, 0,0,0, 0,0,1);

clear;   draw Tower;
برنامهء فرميينی برای تاشه های شبيه به شکل 195
(*) Formian scheme for configurations

   similar to Fig 195:  Conical tower.

   See Section 3.3 of Reference 3. (*)

H=20;   (*) Height of tower (*)

H1=80; (*) Height at the neck of tower (*)

R=8;     (*) Radius at the base of tower (*)

R1=0;   (*) Radius at the neck of tower (*)

m=10;   (*) Number of sides of the base (*)

n=4;      (*) Number of rhombic units in

                               the vertical direction (*)

B=-2*n*H1/H;

P=(R*R-R1*R1)/(H1*H1);

E=rinit(m,n,2,2)|(lamit(0,B+1)|

  [1,0,B; 1,-1,B+1]#{[1,-1,B+1; 1,1,B+1],

  [1,0,B+2; 1,2,B+2]});

Tower=bh(R1,360/(2*m),H/(2*n),P)|E;

use &,vm(2),vt(2),

   vh(10*R,10*R,-6*R, 0,0,0, 0,0,1);

clear;   draw Tower;
برنامهء فرميينی برای تاشه های شبيه به شکل 196
(*) Formian scheme for configurations

   similar to Fig 196:  Conical tower.

   See Section 3.3 of Reference 3. (*)

H=22;   (*) Height of tower (*)

H1=30; (*) Height at the neck of tower (*)

R=15;   (*) Radius at the base of tower (*)

R1=0;   (*) Radius at the neck of tower (*)

m=10;  (*) Number of sides of the base (*)

n=4;     (*) Number of rhombic units in

                              the vertical direction (*)

B=-2*n*H1/H;

P=(R*R-R1*R1)/(H1*H1);

E=rinit(m,n,2,2)|(lamit(0,B+1)|

  [1,0,B; 1,-1,B+1]#{[1,-1,B+1; 1,1,B+1],

  [1,0,B+2; 1,2,B+2]});

Tower=bh(R1,360/(2*m),H/(2*n),P)|E;

use &,vm(2),vt(2),

   vh(10*R,10*R,-2*R, 0,0,0, 0,0,1);

clear;   draw Tower;
برنامهء فرميينی برای تاشه های شبيه به شکل 197
(*) Formian scheme for configurations

   similar to Fig 197:  Cylindrical tower.

   See Section 3.3 of Reference 3. (*)

H=20;   (*) Height of tower (*)

H1=20; (*) Height at the neck of tower (*)

R=8;     (*) Radius at the base of tower (*)

R1=8;   (*) Radius at the neck of tower (*)

m=12;  (*) Number of sides of the base (*)

n=4;     (*) Number of rhombic units in

                              the vertical direction (*)

B=-2*n*H1/H;

P=(R*R-R1*R1)/(H1*H1);

E=rinit(m,n,2,2)|(lamit(0,B+1)|

  [1,0,B; 1,-1,B+1]#{[1,-1,B+1; 1,1,B+1],

  [1,0,B+2; 1,2,B+2]});

Tower=bh(R1,360/(2*m),H/(2*n),P)|E;

use &,vm(2),vt(2),

   vh(10*R,10*R,-R, 0,0,0, 0,0,1);

clear;   draw Tower;
برنامهء فرميينی برای تاشه های شبيه به شکل 198
(*) Formian scheme for configurations

   similar to Fig 198:  Lazy-tongs.

   See Section 2.4 of Reference 2. (*)

T=90;  (*) Control angle (*)

D=1.2;  (*) Dimensions

                   of a duplet for T=90 (*)

m=5;    (*) Frequency of duplets (*)

H=sqrt|2*D*sin|(T/2);

V=sqrt|2*D*cos|(T/2);

Fold=rin(1,m,H)|{[0,0,0; H,0,V],

   [0,0,V; H,0,0]};

use &,vn(250,150),vs(35),

   vh(0, m*H,0, 0,0,0, 0,0,1);

clear;   draw Fold;
برنامهء فرميينی برای تاشه های شبيه به شکل 199
(*) Formian scheme for configurations

   similar to Fig 199:  Foldable double layer

   grid. See Section 2.4 of Reference 2. (*)

T=120; (*) Control angle (*)

D=0.8; (*) Dimensions

                            of a duplet for T=90 (*)

m=6;    (*) Frequency in the x-direction (*)

n=5;     (*) Frequency in the y-direction (*)

H=sqrt|2*D*sin|(T/2);

V=sqrt|2*D*cos|(T/2);

Fold=rinid(m,n+1,H,H)|{[0,0,0; H,0,V],

   [0,0,V; H,0,0]}#rinid(m+1,n,H,H)|

   {[0,0,0; 0,H,V], [0,0,V; 0,H,0]};

use &,vn(150,250),vs(35),

   vh(m*H, n*H,15*H, 0,0,0, 0,0,1);

clear;   draw Fold;
برنامهء فرميينی برای تاشه های شبيه به شکل 200
(*) Formian scheme for configurations

   similar to Fig 200:  Foldable double layer

   barrel vault. See Section 2.4 of
   Reference 2. (*)

T=140;  (*) Control angle (*)

L1=0.8; (*) Length of

                            upper part of uniplet (*)

L2=0.7; (*) Length of

                            lower part of uniplet (*)

m=7;     (*) Circumferential frequency (*)

n=5;      (*) Longitudinal frequency (*)

L =L1+L2;   T1=180-T;

D=sqrt|(L1*L1+L2*L2-2*L1*L2*cos|T1);

D1=sqrt|(L*L-D*D);

A=asin|(L2*sin|T1/D);

B=180+A-T;

R=D*sin|B/(sin|B-sin|A);

C=asin|(L*sin|B/R);

E=rinit(m,n+1,C,D1)|{[R-D,0,0; R,C,0],

   [R,0,0; R-D,C,0]}#rinit(m+1,n,C,D1)|

   {[R-D,0,0; R,0,D1], [R,0,0; R-D,0,D1]};

F=bc(1,1,1)|E;   P=m*C/2;

Fold=verad(0,0,90-P)|F;

use &,vn(150,100),vs(30),

   vh(5,8*R,-12*R, 0,0,R, 0,1,R);

clear;   draw Fold;

